

 **EXCEL**

 **WISSEN**

 **FORMELN**

Michael Becker

Vorwort

Immer mal wieder werden Excel-Fragen an mich herangetragen, die ich in dieser Publikation einmal zusammengestellt habe.

Eine grundsätzliche Excel-Bedienung setze ich voraus und deswegen möchte ich auch nicht mehr den Bildschirmaufbau und die Eingabe und Formatierung von Daten besprechen. Der Schwerpunkt soll auf den Formeln liegen. Beginnen werden wir mit ein paar Basics; dann sollen die häufig benutzten Formeln besprochen werden (Seite 11); es folgt ein bisschen Insiderwissen, um das Leben mit Excel angenehmer zu machen (Seite 30); und zu guter Letzt gibt es Beispiele aus der Praxis, in der die Formeln zum Einsatz kommen (Seite 50).

Wer Fragen hat, darf sie mir gerne stellen. Wenn sie von allgemeinem Interesse sind, werde ich sie hier aufnehmen.

Michael Becker

<https://www.bemi-online.de/excel>

Version 8 | April 2024

Copyright © 2018–2024 Michael Becker, <https://www.bemi-online.de>

Dieses Dokument darf in seiner Gesamtheit unverändert gerne weitergegeben werden – vorzugsweise aber als Link zu <https://www.bemi-online.de/excel> für die neueste Version. Bei einem Auszug aus dieser Datei ist der Link zum Original anzubringen.

Darstellung

In den meisten Fällen sind Zellen farblich gekennzeichnet. Dabei entspricht
gelb: eine Benutzereingabe;
blau: ein ermittelter Wert, eine Berechnung durch Excel.

Die Herkunft der Daten in nicht gekennzeichneten Zellen spielt für die Beispiele keine Rolle. Sie könnten zum Beispiel aus einem anderen Programm importiert sein.

	A	B
1	Eingabe:	8
2	Ausgabe:	64

Excel 2021 Eine Funktion, die so gekennzeichnet ist, kann nur mit Excel 2021, einer späteren Version bzw. mit dem Excel aus Microsoft 365 – sofern die neuesten Updates verwendet werden – genutzt werden. Nutzt man eine solche Funktion mit einer älteren Excel-Version, wird es zu einer Fehlermeldung kommen.

INHALT

Grundlagen	6
Zelle, Zeile, Spalte	6
Formeln allgemein	7
Wahrheit und Logik	8
Vergleiche	10
Formeln.....	11
Einfache Berechnungen.....	11
SUMME.....	12
TEILERGEBNIS	13
MIN und MAX, MITTELWERT, ANZAHL.....	14
UND und ODER, oder doch NICHT?	15
WENN	16
ZÄHLENWENN(S), SUMMEWENN(S)	18
SVERWEIS	19
XVERWEIS	21
(AUF-/AB-)RUNDEN, GANZZAHL, ABS	23
SUMMENPRODUKT	25
ZUFALLSZAHL, ZUFALLSBEREICH, ZUFALLSMATRIX.....	26
LINKS, RECHTS, TEIL, LÄNGE, GLÄTTEN	27
VERKETTEN, TEXTKETTE und &.....	28
WERT und TEXT	29
Datum und Zeit.....	30
JAHR, MONAT ..., DATUM und ZEIT.....	31
HEUTE und JETZT	33
ARBEITSTAG und NETTOARBEITSTAGE.....	34
Know-How	35
Ansicht fixieren	35
Bereiche ausfüllen	36
Zellen/Spalten/Zeilen mit \$ fixieren.....	37
Funktionen und Parameter	38
Errechneten Wert fest einfügen.....	41
Automatische Berechnung	43
Einheiten in einer Zelle anzeigen.....	44
Zirkelbezüge.....	45
Pivot-Tabellen.....	46

Aus der Praxis	50
Fremdwährungen umrechnen.....	51
Werte aus einem Text extrahieren.....	52
SVERWEIS mit zwei Indexspalten	54
Datum aus einem Textformat extrahieren.....	56
Duplikate und fehlende Einträge.....	57
Fehler abfangen.....	58
Absatzplanung	61
Gleitender Durchschnitt	64

GRUNDLAGEN

Zelle, Zeile, Spalte

In diesem ersten Kapitel sollen noch einmal die Grundlagen für Formeln besprochen werden. Das Verständnis für Zeilen, Spalten und vor allem für Bereiche ist notwendig für alle weiteren Formeln.

Eine **Zeile** steht für nebeneinander liegende Zellen und wird mit Zahlen unterschieden:

4			
5	Zeile 5	Zeile 5	Zeile 5
6			
7			

Eine **Spalte** steht für untereinander liegende Zellen und wird mit Buchstaben unterschieden:

E	F	G
	Spalte F	
	Spalte F	
	Spalte F	

Und bei mehr als 26 Spalten? Dann geht es mit Doppelbuchstaben weiter: X, Y, Z, AA, AB, AC ... BA, BB, BC ...

Eine **Zelle** besteht aus genau einem Bereich, an dem eine sich eine Spalte mit einer Zeile schneidet. Man gibt immer erst die Spalte, dann die Zeile an, also immer erst Buchstabe, dann Zahl:

	A	B
1	Zelle A1	Zelle B1
2	Zelle A2	Zelle B2

Zellbereiche

Später, wenn wir zu den Formeln kommen, werden Zellbereiche benötigt, denn es könnte recht aufwendig werden, bei einer Summierung hunderte von Zellen von Hand einzugeben. Bei einem Bereich werden die erste und die letzte Zelle angegeben und durch einen Doppelpunkt getrennt. Es ist so, als ob man einem Rahmen um diesen Bereich zieht, von links oben nach rechts unten.

	A	B	C	D
1		Bereich B1:C2		
2				
3				

Möchte man sich nicht auf einen bestimmten Bereich festlegen, sondern komplette Spalten oder Zeilen, kann man umgekehrt die Zeile bzw. Spalte weglassen, zum Beispiel A:C für drei ganze Spalten oder 8:9 für zwei ganze Zeilen.

Formeln allgemein

Eine Formel in Excel hat immer den folgenden Aufbau: Gleichheitszeichen = und Berechnung.

fx =A3+B5*0,3

fx =SUMME(A4:A8)-C7

Das wäre eigentlich schon alles, was man zur Eingabe von Formeln wissen muss. Im Kapitel [Formeln](#) auf Seite 11 geht es dann richtig los. Ich empfehle, aber auch die nachfolgenden Abschnitte durchzulesen, da sie wichtige Grundlagen enthalten.

Apostroph und Textformatierung

Formeln beginnen also mit einem Gleichheitszeichen. Und wenn man dieses Gleichheitszeichen nun als Text benötigt? Ganz einfach: Man stellt ein Apostroph voran.


Ein Apostroph ' ist übrigens auf der üblichen PC-Tastatur das Zeichen über der Raute # (oder Doppelkreuz oder Hash). Die Zeichen ´ und ` sind Akzente und haben hier nichts zu suchen.

Gleiches gilt auch, wenn man mit einem Berechnungszeichen wie + startet.

fx '= hier steht ein Text

fx '+49 7589 415-378

Eine Alternative besteht darin, schon vorab eine Zelle als Text zu formatieren. Dazu die Zelle auswählen, rechte Maustaste drücken und im Kontextmenü *ZELLEN FORMATIEREN ...* auswählen.



Wahrheit und Logik

Einsen und Nullen. So arbeitet ein Computer auf allerunterster Ebene. Aber auch in Excel werden logische Operationen benötigt, insbesondere wenn später über Bedingungen und die WENN-Formel gesprochen wird. Excel zeigt jedoch keine Einsen und Nullen an, sondern – in der deutschen Version – WAHR (1) und FALSCH (0).

Es wird jetzt etwas theoretischer. Aber das Verständnis für logische Operationen ist wichtig für später. Der Praxisteil kommt dann bei der Besprechung der Formeln. Versprochen.

ODER

Vergleicht man zwei oder mehr Bedingungen miteinander und verknüpft sie mit einer ODER-Operation, dann ist das Gesamtergebnis WAHR, wenn nur eine der Bedingungen WAHR ist. Daraus ergibt sich die folgende Logik-Tabelle:

Bedingung 1	Bedingung 2	ODER-Ergebnis
WAHR	WAHR	WAHR
WAHR	FALSCH	WAHR
FALSCH	WAHR	WAHR
FALSCH	FALSCH	FALSCH

Bedingung 1	Bedingung 2	Bedingung 3	ODER-Ergebnis
WAHR	WAHR	WAHR	WAHR
WAHR	FALSCH	FALSCH	WAHR
WAHR	WAHR	FALSCH	WAHR
FALSCH	WAHR	FALSCH	WAHR
...	WAHR
FALSCH	FALSCH	FALSCH	FALSCH

UND

Wenn man nun zwei oder mehr Bedingungen miteinander vergleicht und sie mit einer UND-Operation verknüpft, dann ist das Gesamtergebnis nur WAHR, wenn alle Bedingungen WAHR sind. Daraus ergibt sich die folgende Logik-Tabelle:

Bedingung 1	Bedingung 2	UND-Ergebnis
WAHR	WAHR	WAHR
WAHR	FALSCH	FALSCH
FALSCH	WAHR	FALSCH
FALSCH	FALSCH	FALSCH

Bedingung 1	Bedingung 2	Bedingung 3	UND-Ergebnis
WAHR	WAHR	WAHR	WAHR
WAHR	FALSCH	FALSCH	FALSCH
WAHR	WAHR	FALSCH	FALSCH
FALSCH	WAHR	FALSCH	FALSCH
...	FALSCH
FALSCH	FALSCH	FALSCH	FALSCH

Nun muss man diese Abfrage und Bedingungen in Excel eingeben. Leider geht es nicht à la "Bedingung 1 UND Bedingung 2", sondern man muss sie eingeben wie alle Excel-Funktionen in Form der Funktion und den

Bedingungen als Parameter. Wer an dieser Stelle schon etwas weiterlesen möchte, dem sei das Kapitel [Funktionen und Parameter](#) auf Seite 38 empfohlen.

Die Eingabe sieht wie folgt aus:

in C1:

<i>fx</i>	=ODER(A1="Meier";B1=12)
-----------	-------------------------

in C2:

<i>fx</i>	=UND(A2="Müller";B2=14)
-----------	-------------------------

Ergibt:

	A	B	C
1	Meier	14	WAHR
2	Müller	12	FALSCH

Und wie man sieht, gibt Excel tatsächlich als Lösung WAHR oder FALSCH aus. Das ergibt in den allermeisten Fällen wenig Sinn. Aber später werden wir mit dem Ergebnis arbeiten und nicht einfach nur ausgeben.

Es gibt noch weitere solcher Logik-Operationen, aber für den Anfang soll das erst einmal ausreichend sein.

Zu dem Beispiel oben noch eine Anmerkung: Texte werden in Anführungszeichen gesetzt.

Vergleiche

Immer wieder müssen zwei Werte miteinander verglichen werden. Einen Vergleich haben wir bereits im vorherigen Kapitel gesehen, nämlich die Gleichheit von zwei Werten: $A1="Meier"$.

Die weiteren Vergleiche werden in der gleichen Art und Weise geschrieben:

Vergleich	Erklärung	Beispiel
>	größer als	$5>4$ ergibt WAHR
>=	größer als oder gleich	$3>=3$ ergibt WAHR
<	kleiner als	$9<7$ ergibt FALSCH
<=	kleiner als oder gleich	$6<=6$ ergibt WAHR
<>	ungleich	$8<>8$ ergibt FALSCH

Auch wenn es in Excel erst einmal etwas merkwürdig aussieht, aber wir erinnern uns: Jede Formel beginnt mit einem Gleichheitszeichen. Wir schreiben in B1:

<i>fx</i>	=A1>=A2
-----------	---------

Ergibt:

	A	B
1	4	WAHR
2	3	

Im Übrigen können auch Texte miteinander verglichen werden, und zwar einfach anhand ihrer Position im Alphabet: "Koch"<"Müller", "Schulze">"Meier".

Öfters kommt die Frage, welches denn nun das richtige Zeichen ist. Merkregel 1: Auf der Seite, an der das Zeichen auseinander geht, steht der größere Wert, an der kleinen Spitze der kleinere Wert. Größer > kleiner, kleiner < größer. Merkregel 2: Denkt man sich einen senkrechten Strich neben dem Zeichen, bekommt man ein K wie kleiner: |<.

FORMELN

Einfache Berechnungen

Die Berechnung beginnt wie immer mit einem Gleichheitszeichen. Danach können alle möglichen Berechnungen durchgeführt werden, egal ob mit Zellbezügen oder mit festen Zahlen. Es gilt die Punkt- vor Strichrechnung. Und die Klammersetzung ist ebenfalls möglich.

<i>fx</i>	=0,7+0,3*B5
-----------	-------------

<i>fx</i>	=(C2-C1)*2/D4
-----------	---------------

<i>fx</i>	=A1*(1+B1/100)^C1
-----------	-------------------

Letzteres ist übrigens die allgemeine Zinseszinsformel, wenn A1 das Grundkapital ist, B1 der Prozentsatz und C1 die Anzahl der Jahre. Das Zeichen ^ steht für die Potenz. 5^3 entspricht in Excel 5^3 .

Auch eine Kombination mit Formeln (mehr dazu in den folgenden Kapiteln) ist selbstverständlich möglich:

<i>fx</i>	=SUMME(D1:D5)*0,7+SUMME(E1:E5)
-----------	--------------------------------

Und da man es leider immer wieder sieht: Formeln wie $=+A3+B4$ oder $=(A1+B3+C5)$ sind mathematisch einwandfrei und werden auch so berechnet. Nur schön sieht es nicht aus. Und bei komplexeren Formeln kann es zu Verwirrungen führen.

Noch ein Wort zu Prozentrechnungen: Excel kann mit % umgehen. Eine Eingabe von 3% innerhalb einer Formel wird korrekt als 0,03 interpretiert. % kommt aus dem Lateinischen und heißt "pro hundert". Es hat sich im allgemeinen Sprachgebrauch eingebürgert, dass man sagt: 50 EUR + 19 % MwSt. Jeder weiß, was gemeint ist, aber mathematisch kommt hier 50,19 heraus (die Mathematiker verzeihen mir die unterschlagene Einheit). Richtig wäre 50 EUR + 19 % von 50 EUR. Oder in der Excel-Schreibweise: $=50*1,19$ oder $=50*(1+19\%)$.

Somit könnte die Zinseszinsformel oben auch lauten: $=A1*(1+B1\%)^C1$.

SUMME

Ganz am Anfang steht geschrieben, dass es neben Zeilen, Spalten und Zellen auch Zellbereiche gibt. Und die sind ganz wichtig für Funktionen wie die SUMME-Funktion. Sie eignet sich dafür, wenn mehrere Werte, die untereinander oder nebeneinander oder sogar unter- und nebeneinander stehen, aufsummiert werden sollen. Es lassen sich sogar Bereiche addieren, die nicht zusammenhängend sind. Dazu später mehr im Kapitel [Funktionen und Parameter](#) auf Seite 38.

Rufen wir uns noch einmal ins Gedächtnis: Ein Zellbereich beginnt mit der ersten Zelle (links oben), dann steht ein Doppelpunkt, und es folgt die letzte Zelle (rechts unten).

Hier ein Beispielblatt:

	A	B	C	D	E	F
1		1. Quartal	2. Quartal	3. Quartal	4. Quartal	Summe
2	Deutschland	500	450	520	490	1960
3	Frankreich	200	210	220	240	870
4	Belgien	200	180	190	180	750
5	Summe	900	840	930	910	3580

Und das steht in den einzelnen Zellen:

In B5:

<i>fx</i>	=SUMME(B2:B4)
-----------	---------------

In F2:

<i>fx</i>	=SUMME(B2:E2)
-----------	---------------

In F5:

<i>fx</i>	=SUMME(B2:E4)
-----------	---------------

In F5 wäre natürlich auch eine Summe der Quartalssummen oder der Ländersummen möglich gewesen.

Auch hier gibt es wieder unschöne Ausdrücke, die zwar richtig berechnet, aber am besten ganz schnell vergessen werden, zum Beispiel =SUMME(A1+A2+A3) oder =SUMME(A1*5).

TEILERGEBNIS

	A	B	C	D	E	F
1	Gruppe ▼	Land ▼	Q1 ▼	Q2 ▼	Q3 ▼	Q4 ▼
2	Motoren	Schweden	500	450	520	490
5	Pumpen	Schweden	400	470	480	420
8	Elektronik	Schweden	450	450	440	430
11						
12		Summe	2600	2560	2760	2620

Was ist hier passiert? Die Summen passen überhaupt nicht zu den Zahlen darüber.

Die Lösung ist ganz einfach: Es wurde ein Filter auf die Tabelle angewendet. Die SUMME-Funktion berechnet weiterhin die Summen aus dem jeweiligen Bereich darüber, also auch aus den gerade ausblendeten Zeilen.

Und da es häufig passiert, dass man eine Summe bildet und anschließend einen Filter setzt, wird hier die TEILERGEBNIS-Funktion direkt am Anschluss an die SUMME-Funktion erläutert. Man muss leider selbst daran denken. Beim Setzen eines Filters behält Excel die ursprüngliche SUMME-Funktion bei.

Wir schreiben nun einfach in die Zelle unter den Umsatzdaten:

<i>fx</i>	=TEILERGEBNIS(9;C2:C10)
-----------	-------------------------

Der zweite Parameter ist klar. Das ist der Bereich wie bei der Summe auch. Aber an erster Stelle steht eine 9. Und 9 ist lediglich eine Kennzeichnung für die Summe. Wer wissen möchte, was es mit den anderen Kennzeichnungen auf sich hat, der darf versuchen, das mit Hilfe von Seite 38 selbst herauszufinden.

Ein Tipp noch, der generell für gefilterte Tabellen gilt: Beim Eingeben der Formeln immer darauf achten, dass man die gesamte Tabelle erwischt. Am besten dazu anfangs alle Filter entfernen. Ansonsten kann es passieren, dass man einen Zellbereich markiert, aber bei der Auswahl eines anderen Filters Zeilen oben oder unten fehlen. Immer noch einmal schauen, ob der ausgewählte Zellbereich alles umfasst.

MIN und MAX, MITTELWERT, ANZAHL

In diesem Kapitel mache ich es mir recht einfach und schreibe gleich zu mehreren Funktionen etwas. Auf den ersten Blick haben die Funktionen nichts miteinander zu tun. Auf den zweiten Blick schon. Denn sie erwarten alle einen Bereich als Parameter. Also genau wie die SUMME-Funktion, der ich allerdings zum Einstieg ein eigenes Kapitel gewidmet habe.

Wir nehmen die folgende Tabelle zur Hand.

	A	B	C	D	E	F
1	Wert 1	Wert 2	Wert 3	Wert 4	Wert 5	Wert 6
2	4	5	3	8		7
3						
4	Minimum:	3				
5	Maximum:	8				
6	Mittelwert:	5,4				
7	Anzahl:	5				

Die Berechnung findet mit den nachfolgenden, selbsterklärenden Formeln statt.

fx =MIN(A2:F2)

fx =MAX(A2:F2)

fx =MITTELWERT(A2:F2)

fx =ANZAHL(A2:F2)

Und so sind auf die Schnelle gleich mehrere Funktionen gelernt.

Wie man sieht, ist unter Wert 5 nichts eingetragen. Und wer nun einmal nachrechnet oder nachzählt, wird feststellen, dass Excel nur die Zellen berücksichtigt, in denen auch tatsächlich Werte stehen. Bitte in das leere Feld einmal die Zahl null hineinschreiben. Drei der vier Ergebnisse ändern sich. Und was ist nun richtig, eine Null oder ein leeres Feld? Das kommt immer auf den Einzelfall an. Wurde in einem Monat tatsächlich kein Umsatz gemacht, geht auch der Durchschnitt runter. Fehlt ein Messwert in einer Leistungsmessung, sollte man nicht zwangsläufig null als minimalen Messwert ansehen.

UND und ODER, oder doch NICHT?

In den Grundlagen wurden logische Operationen erklärt. Was man damit anfangen kann, wird an dieser Stelle und im nachfolgenden Kapitel über die WENN-Funktion veranschaulicht.

Aus einer recht großen Umsatzliste sollen die Kunden ausgeblendet werden, die in drei Jahren überhaupt keinen Umsatz gemacht haben.

	A	B	C	D	E
1	Kunde ▼	U 2021 ▼	U 2020 ▼	U 2019 ▼	löscher ▼
2	Müller	400	350	450	FALSCH
3	Schmidt	200	100	0	FALSCH
4	Fischer	600	800	750	FALSCH
5	Schulze	0	0	0	WAHR
6	Koch	150	250	250	FALSCH
7	...				

Kein Umsatz in drei Jahren bedeutet: kein Umsatz in 2021 und kein Umsatz in 2020 und kein Umsatz in 2019.

Genau das schreiben wir in Excel in die Zelle E2 und analog in die nachfolgenden Zellen:

```
fx =UND(B2=0;C2=0;D2=0)
```

Excel überprüft nun jede einzelne Bedingung, und wenn alle Bedingungen WAHR ergeben, ist auch das Endergebnis und folglich die Ausgabe WAHR. Nun braucht nur noch ein Filter gesetzt zu werden.

Man kann es aber auch umdrehen und mit ODER formulieren:

```
fx =ODER(B2<>0;C2<>0;D2<>0)
```

Wiederum überprüft Excel jede einzelne Bedingung. Es muss nur in einer der drei Zellen ein Umsatz ungleich null stehen und schon ist das Ergebnis WAHR. Der Filter muss nun aber ebenfalls getauscht werden!

Logische Operationen umkehren

Wie wir gerade gesehen haben, lässt sich eine Bedingung umkehren. In diesem Fall ist es egal, ob man es mit UND oder ODER schreibt; nach dem Filtern hat man in beiden Fällen das gleiche gewünschte Ergebnis.

Bei einfachen Vergleichen ohne Einsatz einer Funktion braucht man nur das Zeichen umdrehen. Möchte man zum Beispiel wissen, ob A1 nicht größer oder gleich A2 ist, dann fragt man: Ist A1 kleiner A2?

Es gibt jedoch noch eine weitere Funktion, die aus WAHR FALSCH und aus FALSCH WAHR werden lässt, nämlich NICHT. Als Parameter zu dieser Funktion übergibt man lediglich den Wahrheitswert, der umgekehrt werden soll.

Und somit bekommt man mit NICHT und der obenstehenden ODER-Formel wieder genau das gleiche Resultat, als hätte man die UND-Funktion genommen:

```
fx =NICHT(ODER(B2<>0;C2<>0;D2<>0))
```

Der ODER-Teil liefert WAHR zurück, aber aus =NICHT(WAHR) wird FALSCH. Klingt logisch, oder?

WENN

In den vorherigen Kapiteln wurde schon einiges über Vergleiche geschrieben, sowohl mit einfachen mathematischen Vergleichen also auch mit den Funktionen UND und ODER. Als Ergebnis zeigt Excel bisher WAHR oder FALSCH an. Nun wollen wir diese Ergebnisse mit der WENN-Funktion weiterverwenden.

Die WENN-Funktion wird benötigt, wenn man ein Ergebnis in Abhängigkeit von anderen Werten ausgeben will. Besonders wichtig ist diese Funktion bei der Bearbeitung von umfangreicheren Listen, denn dann leistet sie eine gute Vorarbeit, um anschließend mit dem Filter arbeiten zu können.

WENN erwartet drei Parameter, die durch ein Semikolon getrennt sind: als erstes eine Bedingung; als zweites einen Wert, wenn die Bedingung WAHR ist (also wenn – dann); und als drittes einen Wert, wenn die Bedingung FALSCH ist (also wenn – sonst). Diese drei Werte dürfen alle beliebigen Bedingungen, Vergleiche und Funktionen enthalten, die wir schon kennengelernt haben

Wir prüfen, ob der Wert in B2 größer null ist. Ist er das, wird der Wert ausgegeben, ansonsten der Wert aus C2:

```
fx =WENN(B2>0;B2;C2)
```

Und etwas umfangreicher: Wenn in A5 "ja" steht, werden die Werte aus Spalte C summiert, sonst aus Spalte D:

```
fx =WENN(A5="ja";SUMME(C2:C17);SUMME(D2:D17))
```

Als Bedingung kann jeder Ausdruck verwendet werden, der als Ergebnis WAHR oder FALSCH liefert:

```
fx =WENN(ODER(A1>0;A2>0);"mind. eine Zahl ist größer null";"keine Zahl ist größer null")
```

Mit der WENN-Funktion lässt sich auch ein Fehler abfangen, zum Beispiel eine Division durch null. (Anmerkung: Es gibt auch andere Möglichkeiten zur Fehlerabfrage (Seite 58), aber hier reicht WENN.)

```
fx =WENN(B2<>0;B1/B2;"ungültig!")
```

Verschachteln von Funktionen

In den Wenn-, Dann- und Sonst-Teil dürfen beliebige weitere Funktionen untergebracht werden. Daher ist es auch erlaubt, mehrere WENN-Funktionen zu verschachteln, das heißt, WENN in den Dann- oder Sonst-Teil einzubringen.

```
fx =WENN(B2>0;B2;WENN(C5=0;C5;B2))
```

Ist die Bedingung B2>0 nicht gültig, dann wird eine weitere Bedingung aufgerufen, nämlich im Sonst-Teil. Ist die erste Bedingung gültig, dann wird der Wert der Zelle C5 nie eine Rolle spielen, da Excel diesen Teil gar nicht überprüft.

Schrittweise aufbauen

Manche WENN-Formeln sind recht einfach: Zwei Werte werden miteinander verglichen. Im Erfolgsfall wird das eine ausgegeben, ansonsten das andere.

So einfach ist es leider nicht immer. Man stelle sich eine große Tabelle mit Produkten, den dazugehörigen technischen Daten sowie den passenden Kunden mit kaufmännischen Daten dazu vor. Es soll nun geprüft

werden, ob dieses Produkt für den betroffenen Kunden unter eine Richtlinie fällt und ob das Produkt überhaupt noch geliefert wird.

	A	B	C	D	E	F	G	H	I	J
1	Produkt	Wert 1	Wert 2	Wert 3	...	Kunde	Land	Umsatz	...	betroffen?
2	ABC-45	120	0,3	80	...	Smith	USA	80000	...	ja
3	...									

Die dazugehörige Formel in J2 und folgenden Zellen analog lautet:

<i>fx</i>	=WENN(UND(ODER(B2>=75;C2>0);D2<90;G2="USA";H2>0);"ja";"nein")
-----------	---------------------------------------------------------------

Sieht in der fertigen Version zunächst etwas verwirrend aus. Aber eine solche Formel lässt sich sowohl besser verstehen als auch aufbauen, wenn man sie in kleine Stücke zerlegt. Beim Verstehen würde ich von innen nach außen vorgehen. In diesem Fall sind der Dann- und der Sonst-Teil mehr als einfach gehalten; es interessiert hier also nur die Bedingung. Der innerste Bereich ist derjenige, dessen Klammern zuletzt geöffnet wurden, also hier der ODER-Teil. Betrachtet man nur diesen einen Teil, sieht es gar nicht mehr so schlimm aus. Lediglich zwei Bedingungen, von denen eine WAHR sein muss. Und eine Stufe höher steht der UND-Teil. Auch dieser Teil verliert seinen Schrecken, denn es sind lediglich vier ganz einfache Bedingungen: der ODER-Teil, den wir soeben betrachtet haben, und drei weitere Vergleiche. Es müssen alle vier Bedingungen WAHR sein, damit "ja" ausgegeben wird.

Andersherum, wenn man eine solche Formel aufbauen möchte, ist es nicht viel schwieriger. Wer in ganz kleinen Schritten vorgehen möchte, der fügt sich eine Handvoll Hilfsspalten ein: erste Bedingung, zweite Bedingung etc. So kann man selbst sehr leicht sehen, ob zumindest diese Unterergebnisse schon einmal den erwarteten Ergebnissen entsprechen. Erst dann geht man hin und kombiniert die Bedingungen. Im obigen Beispiel könnte man erst einmal in eine Spalte =B2>75 und in die andere =C2>0 schreiben, und dann stichprobenartig prüfen. Ist alles in Ordnung? Dann werden diese Teile in die nächste Funktion gepackt: =ODER(B2>=75;C2>0); und wieder stichprobenartig geprüft. Und so weiter ...

Ein Tipp noch: Die Anzahl der geöffneten Klammern muss natürlich der Anzahl der geschlossenen Klammern entsprechen. Es können nie mehr Klammern geschlossen werden, als geöffnet wurden.

Vorher denken

Oftmals kommt bei aufwendigeren Formeln schnell Frust auf, wenn gar nichts läuft. Und egal, wie man die Formel modifiziert, es passt einfach nicht. Im Gegenteil, es wird noch schlimmer. Dabei sollte man sich immer bewusst machen, dass Excel nur genau das errechnet oder prüft, was man eingegeben hat. Die Logik, mit der Excel arbeitet, ist nicht zu bestreiten.

Vielmehr liegt es daran, dass man sich vorher gar nicht richtig überlegt hat, was man Excel überhaupt sagen möchte. Man könnte sich zum Beispiel selbst aufschreiben, was genau geprüft und berechnet werden soll und übersetzt dann quasi die deutsche Anweisung in die Schreibweise von Excel. "Wenn Wert 1 größer/gleich 75 ist oder Wert 2 größer als 0, und der Wert 3 kleiner als 90 und der Kunde aus den USA kommt und einen Umsatz größer als 0 macht, dann gebe 'ja' aus, ansonsten 'nein'."

ZÄHLENWENN(S), SUMMEWENN(S)

Bleiben wir doch gleich mal beim WENN. Im vorherigen Kapitel wurde eine bestimmte Bedingung überprüft und die Zelle in Abhängigkeit vom Ergebnis gefüllt. Excel bietet zwei weitere Funktionen an, die sich nicht auf eine bestimmte Bedingung beziehen, sondern es lässt sich ein größerer Bereich abfragen.

	A	B	C
1	Artikel	Lager	Bestand
2	4786	Hamburg	7
3	5371	Hannover	0
4	1824A	Hamburg	11
5	4786	Bremen	6
6	1824B	Bremen	12
7

Um eine einfache Anzahl zu bestimmen, zum Beispiel, wie oft ein Artikel vorkommt, verwendet man:

```
fx =ZÄHLENWENN(A:A;4786)
```

Der erste Parameter ist der Bereich, der geprüft werden soll, und der zweite Parameter, wonach gesucht werden soll. In diesem Fall ist es ein bestimmter Wert, aber man sogar eine Bedingung eingeben und in Anführungszeichen setzen, zum Beispiel ">1000".

Fügt man ein "S" an die Formel an, kann man nach mehreren Bedingungen suchen. Die Bedingungen werden kombiniert. Möchte man wissen, wie viele Artikel in einem Lager liegen, kann man es so abfragen:

```
fx =ZÄHLENWENNS(B:B;"Bremen";C:C;">0")
```

Wichtig: Es müssen hier beide Bedingungen erfüllt sein, also das Lager Bremen und der Bestand größer null. Und wenn jeweils nur eine Bedingung erfüllt sein muss, also eine Oder-Verknüpfung? Ganz einfach: Man zählt sie einzeln zusammen: =ZÄHLENWENN(...)+ZÄHLENWENN(...).

Möchte man nicht nur zählen, sondern auch gleich die Summe bestimmen, dann verwendet man:

```
fx =SUMMEWENN(A:A;4786;C:C)
```

Hier kommt ein dritter Parameter hinzu. Vorne stehen wiederum der Suchbereich und das Suchkriterium. Nun kommt allerdings noch der Bereich hinzu, aus dem summiert werden soll. Falls es keine kompletten Spalten sind, bitte aufpassen, dass die Bereiche mit der gleichen Zeile anfangen, sonst kann Excel natürlich nicht ordentlich zählen, zum Beispiel nicht A2:A6;...;C3:C7.

Analog zu ZÄHLENWENNS gibt es auch die Funktion SUMMEWENNS. Die Bedingungen werden kombiniert und die zugehörigen Werte summiert. Aber Achtung: Da es beliebig viele Bedingungen geben kann, steht der zu summierende Bereich nun vorne!

```
fx =SUMMEWENNS(C:C;B:B;"Hamburg";C:C;">=10")
```

Auch mit Texten können all diese Funktionen umgehen. Hier einmal ein Sternchen als Platzhalter, um die Bestände aller Artikel beginnend mit 1824 zu summieren:

```
fx =SUMMEWENN(A:A;"1824*";C:C)
```

SVRWEIS

Der SVRWEIS ist ein sehr mächtiges Werkzeug, um Daten zu verknüpfen. Diese Funktion kann aus einem Datenbestand sehr schnell einen Wert anhand eines Suchwerts zurückliefern. Man könnte im Zusammenhang mit der SVRWEIS-Funktion schon fast von einer kleinen Datenbank sprechen.

	A	B	C	D	E	F	G
1	Artikelinfo			Artikelnr.	Bezeichnung	Status	Preis
2				10397	XAK-662	gültig	27,18
3	Artikelnr.			10398	YNU-694	gültig	12,28
4	10398			10399	XPB-715	gültig	22,31
5				10400	QHR-829	gelöscht	25,26
6	Bezeichnung:			10401	MVH-550	gültig	17,75
7	YNU-694			10402	LHT-540	gültig	18,29
8				10403	EGO-215	Auslauf	24,63
9	Preis:			10404	TGL-423	gültig	28,62
10	12,28			10405	EEB-502	gültig	26,24

Auf der linken Seite wollen wir uns zu einem Artikel mit der Nummer in Zelle A4 die Bezeichnung und den Preis raussuchen lassen. Im Moment bitte erst einmal darauf beschränken, nur eine existierende Artikelnummer einzugeben. Auf der rechten Seite stehen die Daten. Aber auch mit einer sehr viel größeren Auswahl kommt diese Funktion locker zurecht. Zu beachten ist, dass der Wert, nach dem gesucht werden soll (quasi der Index), immer links von den anderen Werten steht, da wir gleich eine Spaltennummer angeben müssen

In A7 kommt die folgende Funktion hinein:

```
fx =SVERWEIS(A4;D2:G10;2;FALSCH)
```

Die Funktion erwartet vier Parameter. Der erste Parameter ist der gesuchte Wert (Artikelnummer). Der zweite Parameter ist der Bereich, in dem gesucht werden soll, also unser kompletter Artikelbestand samt den zugehörigen Daten. Im dritten Parameter gibt man die Spalte innerhalb dieses Bereichs an. In diesem Fall steht die Bezeichnung in der zweiten Spalte des Bereichs. Und mit dem letzten Parameter gibt man an, ob eine ungefähre Suche erlaubt sein soll. Das ergibt in vielen Fällen nicht viel Sinn, deswegen FALSCH.

Die Formel für den Preis, also in Zelle A10, muss entsprechend lauten:

```
fx =SVERWEIS(A4;D2:G10;4;FALSCH)
```

Vorsicht beim Kopieren. Excel passt alle Zellbezüge entsprechend an. Für Abhilfe empfehle ich die Seite [37](#).

Daten auslagern

Ob nun aus optischen oder aus praktischen Gründen, es bietet sich an, den Datenbestand auszulagern. Die erste Möglichkeit wäre, die Daten in einem anderen Tabellenblatt (Register) zu speichern; bei der zweiten Möglichkeit sogar in einer anderen Datei.

Nehmen wir einmal an, die oben angezeigten Daten befinden sich in der gleichen Datei in einem zweiten Tabellenblatt, beginnend bei A1. Dann sieht die Formel für die Bezeichnung (A7) wie folgt aus:

```
fx =SVERWEIS(A4;Tabelle2!A2:D10;2;FALSCH)
```


Man sieht hier, dass dem Zellbezug der Tabellenname gefolgt von einem Ausrufezeichen vorangestellt ist. Übrigens: Der Tabellenname steht in einfachen Anführungszeichen, wenn ein Leerzeichen enthalten ist, zum Beispiel 'Tabelle 2'!A1. Aber keine Sorge, auch das macht Excel von sich aus. Man kann beim Eingeben der Formel ebenso in einem anderen Tabellenblatt einen Bereich markieren.

Nun bringen wir den Datenbestand in eine komplett andere Datei und bekommen diese Formel:

```
fx =SVERWEIS(A4;[Mappe2.xlsx]Tabelle1!$A$2:$D$10;2;FALSCH)
```

Jetzt steht nicht nur der Tabellenname vor dem Zellbezug, sondern auch noch der Dateiname in eckigen Klammern. Wie gesagt, einfach den Bereich markieren und Excel kümmert sich selbst um die entsprechende Schreibweise.

Excel führt ständig neue Berechnungen durch, und wenn sich ein Wert im Datenbestand ändert, passt sich auch das Ergebnis vom SVERWEIS an. Was aber passiert, wenn man den Datenbestand (Mappe2.xlsx) ändert und die ursprüngliche Datei (Mappe1.xlsx) gar nicht geöffnet ist? Excel merkt sich erst einmal den letzten Wert. Beim Öffnen von Mappe1.xlsx kommt dann die folgende Meldung und Excel kann nach Anklicken von **AKTUALISIEREN** den neuen Wert herauslesen:



NV wie "nicht vorhanden"

Zum Abschluss wollen wir uns noch um eine Fehlermeldung kümmern. Sucht man nach einer Artikelnummer, die gar nicht vorhanden ist, gibt Excel lediglich die Fehlermeldung #NV anstelle des Werts (Bezeichnung oder Preis) aus. Für wen das optisch nicht ansprechend ist, der kann die komplette Formel in einer Fehlerabfrage integrieren.

```
fx =WENNNV(SVERWEIS(A4;D2:G10;2;FALSCH);"UNGÜLTIG!")
```

Mehr dazu gibt es auf gibt es in dem Kapitel [Fehler abfragen](#).

XVERWEIS

Excel 2021

XVERWEIS ist eine Erweiterung von SVERWEIS und bügelt ein paar Unzulänglichkeiten aus: SVERWEIS konnte nur Daten rechts der Suchspalte zurückgeben; XVERWEIS sucht in alle Richtungen, auch dann, wenn die Werte nicht unter-, sondern nebeneinanderstehen. SVERWEIS zeigt einen Fehlercode an, wenn ein Wert nicht gefunden wurde; XVERWEIS liefern in diesem Fall einen vordefinierten Wert zurück. Beim SVERWEIS ist es schnell passiert, dass der letzte Parameter FALSCH für die genaue Suche vergessen wurde, der aber in den allermeisten Fällen unerlässlich war; bei XVERWEIS ist dieser Wert nun Standard.

Wir nehmen noch einmal die Tabelle aus dem SVERWEIS-Beispiel, allerdings vertauschen wir die Spalten mit der Artikelbezeichnung und Artikelnummer.

	A	B	C	D	E	F	G
1	Artikelinfo			Bezeichnung	Artikelnr.	Status	Preis
2				XAK-662	10397	gültig	27,18
3	Artikelnr.			YNU-694	10398	gültig	12,28
4	10398			XPB-715	10399	gültig	22,31
5				QHR-829	10400	gelöscht	25,26
6	Bezeichnung:			MVH-550	10401	gültig	17,75
7	YNU-694			LHT-540	10402	gültig	18,29
8				EGO-215	10403	Auslauf	24,63
9	Preis:			TGL-423	10404	gültig	28,62
10	12,28			EEB-502	10405	gültig	26,24

XVERWEIS bekommt bis zu sechs Parameter, wobei nur die ersten drei wirklich benötigt werden: Suchkriterium, Suchbereich, Rückgabebereich, alternativer Rückgabewert, Vergleichsmodus, Suchmodus. Die letzten drei Parameter dürfen weggelassen werden.

In A7 schreiben wir:

```
fx =XVERWEIS(A4;E2:E10;D2:D10;"UNGÜLTIG!")
```

Es soll also nach der Artikelnummer aus Spalte A4 gesucht werden und zwar in der Spalte E. Der dazugehörige Wert steht in Spalte D, also links davon. Kein Problem. Und auch den Fehler müssen wir nicht mehr mit der WENNV-Funktion abfangen.

Demensprechend kommt in A10:

```
fx =XVERWEIS(A4;E2:E10;G2:G10;"UNGÜLTIG!")
```

Horizontale Suche

Wie schon zuvor geschrieben, kann XVERWEIS nicht nur innerhalb einer Spalte, sondern auch in einer Zeile suchen. Das müssen wir der Funktion nicht einmal mitteilen. Wir geben einfach die beiden Bereiche an.

	A	B	C	D	E	F
1	Januar	Februar	März	April	Mai	Juni
2	31	28	31	30	31	30

Die Suche sieht dann folgendermaßen aus:

fx =XVERWEIS("Februar";A1:F1;A2:F2)

Vergleichsmodus und Suchmodus

Fehlen noch die letzten beiden Parameter. In den vorherigen Beispielen wurden sie einfach weggelassen. Und damit wurde eine exakte Suche vom ersten zum letzten Wert hin durchgeführt.

Der fünfte Parameter steht für den Vergleichsmodus. Hier kann angegeben werden: 0 = exakte Suche (gilt auch, wenn der Parameter fehlt); -1 = exakt oder der nächst kleinere Wert; 1 = exakt oder der nächst größere Wert; 2 = Platzhaltersuche. Es dürfen hier auch die bekannten Joker ? (Einzelzeichen) oder * (mehrere Zeichen) verwendet werden.

Und schließlich steht der sechste und letzte Parameter für die Suchrichtung: 1 = aufsteigend (dies ist der Standard); -1 = absteigend; 2 = aufsteigend in einer sortierten Liste; -2 = absteigend in einer sortierten Liste.

	A	B	C	D	E	F
1	April	Mai	Juni	Juli	August	September
2	30	31	30	31	31	30

Und nun suchen wir die Arbeitstage in den beiden Monaten, die mit "ju" beginnen, einmal vorne (30 Tage) und einmal von hinten (31 Tage):

fx =XVERWEIS("ju*";A1:F1;A2:F2;;2)

fx =XVERWEIS("ju*";A1:F1;A2:F2;;-1)

Mehrere Rückgabewerte

Und nun kommt noch der Clou: Man kann mehrere Werte auf einmal zurückgeben, wofür man früher mehrere Zellen mit einem SVVERWEIS brauchte.

	A	B	C	D	E
1	Umsatz	Q2	Q2	Q3	Q4
2	Meier	200	300	250	350
3	Berger	600	600	700	650
4	Renner	550	450	500	500
5					
6		600	600	700	650

Nur in Zelle B6 kommt nun die nachfolgende Formel. Wie man sieht, ist der Rückgabebereich größer als der Suchbereich. Excel füllt daher die Zellen C6, D6 und E6 mit aus.

fx =XVERWEIS("Berger";A2:A4;B2:E4)

(AUF-/AB-)RUNDEN, GANZZAHL, ABS

Und wieder einmal können wir gleich mehrere Funktionen auf einmal lernen. Wenn man eine Zahl runden möchte, dann gibt man lediglich diese Zahl sowie durch ein Semikolon getrennt die Anzahl der Nachkommastellen an. Letztere kann auch null sein. Excel kann Zahlen auf drei Arten runden.

Abrunden: Die Zahlen werden immer abgerundet:

fx =ABRUNDEN(A2;1)

Aufrunden: Die Zahlen werden immer aufgerundet:

fx =AUFRUNDEN(D2;0)

Runden: Die Zahlen werden immer kaufmännisch gerundet, d. h. ab 5 hinauf und unter 5 hinab:

fx =RUNDEN(G2;2)

Diese drei Formeln ergeben hier:

	A	B	C	D	E	F	G	H
1	Zahl	abrunden		Zahl	aufrunden		Zahl	runden
2	3,28	3,2		37,4	38		1,436	1,44

Anzeige von Nachkommastellen

Wir stellen unserem Kunden eine Rechnung aus.


	A	B	C	D	E
1	Artikel	Menge	Einheit	Einzelpreis	Gesamtpreis
2	Schlauch	1,7	m	3,82	6,49
3					
4				Netto	6,49
5				MwSt	1,23
6				Brutto	7,73

Auch ohne Taschenrechner fällt schnell auf, dass sich hier ein Rundungsfehler eingeschlichen hat.

Excel bietet zwei Schaltflächen an, um mit einem Klick eine Nachkommastelle hinzuzufügen oder zu entfernen:



Alternativ kann man bei der Zellenformatierung direkt eine bestimmte Anzahl von Nachkommastellen eingeben:



Das ist jedoch kein Runden! Im Hintergrund rechnet Excel mit der nicht gerundeten Zahl weiter. In manchen Fällen ist es erwünscht oder sogar notwendig. Bei Beträgen rechnet man üblicherweise mit gerundeten Werten weiter. In diesem Beispiel hätte ich sowohl die Berechnung des Gesamtwerts als auch die Berechnung der Mehrwertsteuer in eine RUNDEN-Funktion gepackt. Also immer dort, wo bei einer Punktrechnung eine höhere Anzahl von Nachkommastellen möglich ist; bei der Strichrechnung spielt es keine Rolle, denn die Anzahl der Nachkommastellen ändert sich nicht.

Ganzzahl

Die GANZZAHL-Funktion rundet immer auf eine Zahl ohne Nachkommastellen ab. Hätte man nicht auch =ABRUNDEN(Zahl;0) nehmen können? Nicht ganz. Bei negativen Zahlen verhalten sich die beiden Funktionen anders. =GANZZAHL(-5,2) und =GANZZAHL(-5,7) ergeben -6; =ABRUNDEN(-5,2;0) und =ABRUNDEN(-5,7;0) ergeben -5.

Absolut

Gewissermaßen wird bei der ABS-Funktion auch etwas weggelassen, und zwar das negative Vorzeichen. ABS liefert immer eine positive Zahl. Sowohl bei =ABS(-3) also auch bei =ABS(3) ist das Ergebnis 3. Die Abkürzung stammt aus der Mathematik. Das Ergebnis wird Absolutwert oder absoluter Betrag genannt.

Auf volle Hunderter runden

Die vorherigen Beispiele haben sich mit Nachkommastellen beschäftigt. Aber wie sieht es aus, wenn man auf volle Hunderter (oder auch Zehner, Tausender ...) runden möchte? Manchmal spielen exakte Beträge für eine Darstellung nur eine untergeordnete Rolle.

Wie wird nun aus 12.428 Stück 12.400 Stück? Man teilt erst einmal durch 100 und bekommt eine Zahl mit Nachkommastellen (124,28), die man dann ohne Nachkommastelle runden kann (124). Anschließend muss man nur noch wieder mit 100 multiplizieren (12.400).

```
fx =RUNDEN(B3/100;0)*100
```

Nun hat sich Excel aber etwas ausgedacht, was auf den ersten Blick etwas unverständlich ist, aber den gleichen Zweck erfüllt. Man gibt eine negative Zahl zum Runden an. Für das Runden auf Hunderter geht man dieses Mal vom Komma aus zwei Stellen nach links.

```
fx =RUNDEN(B3;-2)
```


SUMMENPRODUKT

Die SUMMENPRODUKT-Funktion vereint zwei Rechenschritte in einem. Erst werden in einem Bereich zwei Werte miteinander multipliziert und anschließend diese Zwischenergebnisse aufsummiert. Wir nehmen folgende Tabelle:

	A	B
1	5	2
2	7	3
3	6	4
4		
5		55

Um zu dem Summenprodukt zu kommen, gibt es zwei Varianten: Erstens, wir verwenden eine Hilfsspalte und addieren die Werte. Dann steht in der dritten Spalte 10, 21, 24 und darunter als Summe 55. Wenn man die Zwischenergebnisse sowieso ausweisen muss, ist das schon gar keine schlechte Variante. Zweitens, man berechnet alle Werte in einer einzigen Zelle: $=A1*B1+A2*B3+A3*B3$. Das sieht gut aus. Wird aber Schreibarbeit bei vielen Zeilen.

Alternativ dazu kann man das Summenprodukt verwenden. Man übergibt die Zellbereiche mit den beiden Spalten, die logischerweise die gleiche Größe aufweisen müssen:

<i>fx</i>	<code>=SUMMENPRODUKT(A1:A3;B1:B3)</code>
-----------	------------------------------------------

Und wenn man sich die Formel so anschaut, könnte man der Meinung sein, dass die Werte nicht unbedingt nebeneinanderstehen müssen, vielleicht auch in zwei verschiedenen Blättern, und dass man möglicherweise auch mehr als zwei Bereiche multiplizieren kann. Die Antwort lautet ganz einfach: Ja, das geht.

SUMMENPRODUKT ist eine Matrix-Funktion und bietet sehr viel mehr als hier beschrieben. Dafür ist aber ein grundsätzliches Verständnis für Arbeiten mit einer Matrix notwendig, was leider hier den Rahmen sprengen würde.

ZUFALLSZAHL, ZUFALLSBEREICH, ZUFALLSMATRIX

Viele Einsatzbereiche wird es hierfür nicht geben: die Zufallszahlen. Genau, Excel kann nicht nur Berechnungen anstellen, sondern auch zufällige Zahlen ausgeben.

Für alle nachfolgenden Funktionen ist es wichtig zu wissen, dass ständig neue Zufallszahlen ermittelt werden, sobald eine Neuberechnung durchgeführt wird. Das heißt, selbst eine einfache Eingabe in einer anderen Zelle führt zu einer Neuberechnung und somit zu einer neuen zufälligen Zahl. (Einen Hinweis zu der automatischen Neuberechnung gibt es auch noch auf Seite [43](#).)

Die einfachste Funktion gibt einen Wert größer/gleich 0 und kleiner 1 aus und benötigt keinen Parameter:

fx =ZUFALLSZAHL()

Möchte man nun nicht eine Zahl mit vielen Nachkommastellen, bietet Excel eine weitere Möglichkeit. ZUFALLSBEREICH bekommt die Unter- und Obergrenze übergeben. Für einen regulären Würfel schreibt man:

fx =ZUFALLSBEREICH(1;6)

Gleiches Ergebnis hätte man übrigens auch mit erster Funktion bekommen:

fx =GANZZAHL(ZUFALLSZAHL()*6+1)

Excel 2021 Die neue Funktion ZUFALLSMATRIX bietet die Möglichkeit, viele Zufallszahlen über einen größeren Bereich hinweg zu erzeugen.

fx =ZUFALLSMATRIX(7;4;100;999;WAHR)

Mit dieser Funktion werden auf einen Schlag sieben Zeilen und vier Spalten mit ganzen Zahlen (dafür der letzte Parameter WAHR) zwischen 100 und 999 gefüllt.

Zu guter Letzt sollte man noch im Hinterkopf behalten, dass bei mehreren zufälligen Werten in einem Tabellenblatt Duplikate entstehen können. Jede Funktion in einer Zelle arbeitet für sich. Bei sechs Zufallszahlen für den nächsten Lottoschein könnte eine Zahl durchaus mehrfach auftauchen. In dem Fall einfach F9 drücken.

LINKS, RECHTS, TEIL, LÄNGE, GLÄTTEN

Bis jetzt haben sich die Funktionen vielfach um Zahlen gedreht. Aber Excel bietet auch eine ganze Reihe von Funktionen an, die sich mit Texten beschäftigen. Darum soll es in diesem und den nachfolgenden Abschnitten gehen.

fx =LINKS(A1;5)

fx =RECHTS(B1;3)

Das ist noch recht einleuchtend: Aus den beiden Zellen sollen entweder die ersten fünf Zeichen von links bzw. die letzten drei Zeichen von rechts extrahiert werden.

Möchte man nur einen Teil haben, muss man Excel sagen, an welcher Position begonnen und wie lang der Ausschnitt sein soll, zum Beispiel fünf Zeichen ab/einschließlich der dritten Stelle:

fx =TEIL(C1;3;5)

Um herauszufinden, wie lang ein Text ist, verwendet man:

fx =LÄNGE(A1)

Im Praxisteil auf Seite 51 kommen diese Funktionen auch noch einmal zum Einsatz.

Es kann bei importierten Daten oder Benutzereingaben passieren, dass ein Text (vorzugsweise am Ende) Leerzeichen enthält. Mit der GLÄTTEN-Funktion werden die Leerzeichen am Anfang und am Ende gelöscht.

fx =GLÄTTEN(A1)

Wenn die Wahrscheinlichkeit hoch ist, dass sich Leerzeichen eingeschlichen haben, kann man auf Nummer sicher gehen und die Leerzeichen grundsätzlich vorher entfernen lassen. Wenn beispielsweise "Frank Müller " (also mit Leerzeichen am Ende) in einer Zelle steht, bekommt man dennoch so die richtige Länge:

fx =LÄNGE(GLÄTTEN(A1))

VERKETTEN, TEXTKETTE und &

Excel kann auch Texte miteinander kombinieren, und zwar einfach mit Hilfe des &-Zeichens:

	A	B	C	D
1	Vorname	Nachname	PLZ	Ort
2	Markus	Müller	30163	Hannover
3	Martin	Escher	70184	Stuttgart

Den kompletten Namen bekommt man mit:

```
fx =A2&" "&B2
```

Wie man sieht, werden hier drei Textteile kombiniert: der Vorname, ein Leerzeichen und der Nachname. Ohne das Leerzeichen hätte Excel Vor- und Nachname direkt aneinandergesetzt.

Praktischerweise ist Excel in der Lage, Zahlen einfach als Texte anzusehen. Es ist in diesem Fall keine vorherige Umwandlung notwendig.

```
fx =C2&" "&D2
```

Genau genommen, ist die Schreibweise mit dem &-Zeichen nur eine Verkürzung zweier anderer Funktionen. Und genau an dieser Stelle muss man differenzieren.

Die ältere Funktion, die eigentlich nicht mehr genommen werden soll, aber wegen der Kompatibilität noch erhalten geblieben ist:

```
fx =VERKETTEN(C2;" ";D2)
```

Die neue Funktion lautet:

```
fx =TEXTKETTE(C2;" ";D2)
```

Dafür ist die neuere Funktion aber in der Lage, einen Textbereich wie C5:C8 zu verarbeiten.

WERT und TEXT

Um eine Zahl aus einem Text zu extrahieren, benutzt man die WERT-Funktion. Das Ergebnis ist dann auch tatsächlich eine Zahl – und nicht eine Zahl mit Textformat –, mit der man dann auch weiterrechnen kann.

```
fx =WERT(LINKS(A3;5))*1000
```

Das Gegenstück zu der WERT-Funktion ist die TEXT-Funktion. Im vorherigen Kapitel wurde schon gezeigt, dass Excel beim Verketteten automatisch zu einem Text umwandelt. Diese Funktion jedoch kann die Zahl in einem bestimmten Format darstellen.

	A	B	C	D
1	Artikel	Größe	Farbe	Bestellnummer
2	4758	3	17	004758.03.017

Das Ergebnis bekommt man mit dieser Formel:

```
fx =TEXT(A2;"000000")&"."&TEXT(B2;"00")&"."&TEXT(C2;"000")
```

Der zweite Parameter der TEXT-Funktion gibt das Format an, in diesem Fall einer bestimmten Anzahl an Stellen. Woher weiß man das? Ganz einfach: Es ist der gleiche Code, wie er auch beim Formatieren von Zellen verwendet wird. Mehr dazu gibt es in der Know-How-Rubrik auf Seite [44](#).

Datum und Zeit

Excel kann Datum- und Zeitangaben verarbeiten. Aber widmen wir uns erst einmal der Darstellung bzw. Umrechnung. Manchmal kann es sein, dass eine Zelle falsch formatiert ist. Man gibt zum Beispiel das Datum 19.02.2022 ein und plötzlich steht in der Zelle 44611. Da ist aber kein Problem. Über *ZELLEN FORMATIEREN ...* im Kontextmenü mit der rechten Maustaste bzw. im Menüband kann man zwischen der Standardanzeige und einem beliebigen Datumformat (zum Beispiel TT.MM.JJJJ oder JJJJ-MM-TT) wechseln.

44611 entspricht in etwa 122 Jahre und zweieinhalb Monate in Tagen. Denn Excel beginnt seine Zählung im Jahr 1900 und somit hat der 1. Januar 1900 den Wert 0. Frühere Datumsangaben und Berechnungen kann Excel gar nicht oder nur über Umwege verarbeiten.

Und mit diesem Hintergrundwissen ist es auch nicht schwer zu erraten, dass Excel mit Datumsangaben rechnen kann:

	A	B	C	D	E
1	von:	01.01.2022		Start:	01.02.2022
2	bis:	05.02.2022		Dauer:	7
3	Tage:	36		Ende:	07.02.2022

Im ersten Fall wird subtrahiert (+1, da der Bis-Tag inklusive ist), im zweiten Fall wird addiert (-1, da sonst der Tag danach ermittelt würde):

<i>fx</i>	=B2-B1+1
-----------	----------

<i>fx</i>	=E1+E2-1
-----------	----------

Normalerweise erkennt Excel selbständig, ob das Ergebnis eine Zahl oder ein Datum ist. Aber wenn in der Zelle vorher schon einmal etwas anderes gestanden hat und sie anders formatiert war, dann einfach wie oben beschrieben umstellen.

Nicht anders verhält es sich mit Uhrzeiten. Excel verwendet hier einfach die Uhrzeit als Nachkommastelle, sozusagen als Bruchteil eines Tages. 7 Uhr 30 entspricht 0,3125 (= 7,5 geteilt durch 24). Vor dem Komma steht nichts, da hier kein konkretes Datum angegeben wurde. Aber man könnte auch in einer Zelle schreiben: 19.02.2022 07:30. Als Zahl entspricht das 44611,3125.

Bei der Formatierung der Zelle findet man den entsprechenden Eintrag übrigens unter "Benutzerdefiniert"; dieser lautet TT.MM.JJJJ hh:mm. Man könnte auch 44611,3125 mit dem Format TT.MM.JJJJ darstellen, dann wird die Uhrzeit zwar nicht angezeigt, aber im Hintergrund weiterhin mitgeführt.

Rechnen kann man mit Uhrzeiten ebenfalls:

	A	B
1	Schichtbeginn:	16.02.2022 22:00
2	Arbeit (Std:Min):	08:00
3	Pause (Std:Min):	00:45
4	Schichtende:	17.02.2022 06:45

JAHR, MONAT ..., DATUM und ZEIT

Will man aus einem Datum und/oder einer Uhrzeit die einzelnen Bestandteile abrufen, verwendet man die nachfolgenden selbsterklärenden Funktionen.

	A	B	C	D
1	04.01.2022 16:15:32		Jahr:	2022
2			Monat:	1
3			Tag:	4
4			Stunde:	16
5			Minute:	15
6			Sekunde:	32
7			Wochentag:	3

fx =JAHR(A1)

fx =MONAT(A1)

fx =TAG(A1)

Monat und Tag sind jeweils nur noch eine Zahl, deswegen entfällt die führende Null.

fx =STUNDE(A1)

fx =MINUTE(A1)

fx =SEKUNDE(A1)

fx =WOCHENTAG(A1)

Beim Wochentag wird von Sonntag an mit der 1 gezählt. In diesem Fall also 3 für Dienstag. Man könnte noch einen weiteren Parameter für eine andere Zählung angeben.

Es geht aber auch andersherum. Hat man die einzelnen Bestandteile, kann man damit eine Datum- und/oder Zeitangabe erzeugen, mit der Excel arbeiten kann. Das klingt vielleicht erst einmal ungewöhnlich, aber die Angaben könnten zum Beispiel aus einem anderen Programm importiert worden sein und nur als Text vorliegen. Mehr dazu gibt es im Praxisteil auf Seite 56.

	A	B	C	D
1	Jahr:	2022		Datum:
2	Monat:	1		04.01.2022
3	Tag:	4		Uhrzeit:
4	Stunde:	16		16:15:32
5	Minute:	15		komplett:
6	Sekunde:	32		04.01.2022 16:15:32

fx =DATUM(B1;B2;B3)

... in der Reihenfolge Jahr, Monat und Tag.

fx =ZEIT(B4;B5;B6)

Für eine kombinierte Darstellung aus Datum und Zeit müssen die beiden Berechnungen einfach summiert werden, wie es eingangs zu Datum und Uhrzeit schon beschrieben wurde:

fx =D2+D4

Beziehungsweise, wenn man Datum und Uhrzeit nicht einzeln benötigt:

fx =DATUM(B1;B2;B3)+ZEIT(B4;B5;B6)

Und bitte auf eine entsprechende Zellenformatierung achten. Standardmäßig werden die Sekunden in Excel nicht angezeigt.

HEUTE und JETZT

Möchte man das aktuelle Datum bekommen, verwendet man:

<i>fx</i>	=HEUTE()
-----------	----------

Und für die aktuelle Uhrzeit:

<i>fx</i>	=JETZT()
-----------	----------

Zu beachten ist, dass die JETZT-Funktion Datum und Uhrzeit zurückgibt. Mit einer einfachen Zellenformatierung kann man sich aber auch nur die Uhrzeit anzeigen lassen. Wenn man tatsächlich nur die Uhrzeit für weitere Berechnungen benötigt, dann kann man mit dem Wissen aus den vorhergehenden Kapiteln den Wert von HEUTE vom Wert von JETZT abziehen.

Weiterhin ist zu beachten, dass immer die aktuelle Uhrzeit auf die Sekunde genau verwendet wird. Bei jeder Neuberechnung der Arbeitsmappe, also auch wenn nur in einer anderen Zelle ein Wert eingetragen wird, werden Datum und Uhrzeit aktualisiert; genauso wie bei den Zufallszahlen.

ARBEITSTAG und NETTOARBEITSTAGE

Bisher haben wir mit Excel immer Kalendertage addiert oder Kalendertage zwischen zwei Datumsangaben ermittelt. Im beruflichen Alltag ist es oft wünschenswert, nur die Arbeitstage zu berücksichtigen. Da wären zum einen die immer wiederkehrenden Wochenenden, zum anderen die variablen Feiertage, eventuelle Brückentage und möglicherweise noch ein längerer Betriebsurlaub.

Als Beispiel nehmen wir hier einmal die Feiertage (FT) Karfreitag und Ostermontag. Die Wochenenden (WE) müssen nicht erfasst werden.

12.04.	13.04.	14.04.	15.04.	16.04.	17.04.	18.04.	19.04.	20.04.	21.04.	21.04.
Di.	Mi.	Do.	Fr.	Sa.	So.	Mo.	Di.	Mi.	Do.	Fr.
			FT	WE	WE	FT				

	A	B	C	D	E
1	Start:	12.04.2022		frei:	15.04.2022
2	Tage:	5			18.04.2022
3	Ende ohne FT:	18.04.2022			
4	Ende mit FT:	20.04.2022			

Die zugehörige Funktion lautet ARBEITSTAG und benötigt den ersten Tag, die Anzahl der Tage und als freiwilligen Parameter einen Bereich, der alle freien Tage auflistet. Wie auch im ersten Beispiel zu Datum und Uhrzeit muss hier 1 abgezogen werden, da Excel sonst nur die fünf Tage addiert und somit den nachfolgenden Tag ausgibt.

fx =ARBEITSTAG(B1;B2)-1

fx =ARBEITSTAG(B1;B2;E:E)-1

Die Anzahl der Arbeitstage mit und ohne Feiertage lässt sich umgekehrt mit der Funktion NETTOARBEITSTAGE ermitteln. NETTOARBEITSTAGE liefert den gesamten Zeitraum einschließlich des ersten und des letzten Tages zurück, daher ist hier keine Korrektur notwendig.

	A	B	C	D	E
1	Start:	12.04.2022		frei:	15.04.2022
2	Ende:	20.04.2022			18.04.2022
3	Tage ohne FT:	7			
4	Tage mit FT:	5			

fx =NETTOARBEITSTAGE(B1;B2)

fx =NETTOARBEITSTAGE(B1;B2;E:E)


Andere Länder, andere Sitten: Soll das Wochenende nicht auf Samstag und Sonntag liegen, kann man die Funktionen ARBEITSTAG.INTL und NETTOARBEITSTAGE.INTL verwenden und einen weiteren Parameter für das Wochenende angeben. Wer mehr dazu wissen möchte, schaut einmal auf Seite 38.

Tipp: Man kann einen kompletten und immer wieder benötigten Betriebskalender in einer separaten Datei erfassen. Dann muss er nur an einer Stelle gepflegt werden und alle Mitarbeiter könnten darauf zugreifen.

Ansicht fixieren

Gleich der erste Eintrag in dieser Rubrik hat gar nichts mit Formeln zu tun. Aber vielfach wird verkannt, dass man sich gerade bei großen Tabellen das Leben leichter machen kann. Excel kann eine beliebige Anzahl von Spalten und/oder Zeilen fixieren. Das bedeutet, dass diese Spalten und/oder Zeilen beim Scrollen an ihrem Platz bleiben.

Zu finden ist diese Funktion im Menüband *ANSICHT* unter *FENSTER FIXIEREN*.



Dazu wählt man vorab eine bestimmte Zelle und Excel fixiert dann alles links und oberhalb dieser Zelle. Ist also Zelle C2 gewählt, dann bleiben die Spalten A und B stehen sowie die Zeile 1.

Bereiche ausfüllen

Viele wissen, dass man eine Formel in die nachfolgenden Zellen übertragen kann, indem man den kleinen Punkt in der rechten unteren Ecke einer Zeile anklickt, festhält und nach unten zieht.

	A	B	C
1			Summe
2	32	17	49
3	45	15	
4	33	38	
5	15	48	
6	22	16	
7	23	37	
8	41	46	

Mit der Tastatur erreicht man das gleiche Ergebnis, und zwar, indem man einen Bereich markiert und dann die Tastenkombination Strg+U für unten bzw. Strg+R für rechts drückt. Excel füllt nun die Zellen unterhalb bzw. rechts davon mit der Formel aus der ersten Zelle des Bereichs aus.

Zusammen mit Markierung per Tastatur, z. B. Umschalt+Strg+Ende, lässt sich die Formel in einem Bruchteil der Zeit übertragen. Gerade in Tabellen mit mehreren tausenden von Zeilen spart man sich das sekundenlange Runterziehen mit der Maus.

Zellen/Spalten/Zeilen mit \$ fixieren

Wir möchten eine Preiserhöhung durchführen und bereiten die nachfolgende Tabelle vor.

	A	B	C
1	Erhöhung:	3%	
2			
3	Produkt	Preis alt	Preis neu
4	A189	25,30 EUR	26,06 EUR
5	B412	28,90 EUR	
6	C346	34,80 EUR	

In C3 steht:

```
fx =RUNDEN(B4*(1+B1);2)
```

Jetzt würde man normalerweise die Zelle C4 nach unten ziehen, um die Zellen C5 und C6 entsprechend fortzuführen. Das führt jedoch nicht zum gewünschten Erfolg. In C5 steht nun 28,90 EUR, in C6 steht #WERT!.

Beim Kopieren oder beim Füllen von Zellen wendet Excel einen relativen Zellbezug an. Das bedeutet, der Abstand zu der ursprünglichen Zelle wird auf alle Bezüge in der Formel angewendet. Kopiert man C4 auf C5, dann ist der Abstand 0 Spalten und +1 Zeile. In der Formel von oben wird somit B4 zu B5 (was ja auch richtig ist) und B1 zu B2. Letzteres ist natürlich nicht gewollt, denn in B2 steht überhaupt nichts drin.

Es muss also der erste Zellbezug (alter Preis) fortgeführt werden, aber der zweite Zellbezug (Erhöhung) muss immer gleichbleiben. Das erreicht man, indem man ein Dollarzeichen voranstellt:

```
fx =RUNDEN(B4*(1+$B$1);2)
```

Nun funktioniert auch das Runterziehen oder Kopieren der Formeln. B1 bleibt B1.

Je nachdem, in welche Richtung man Formeln weiterführen möchte, untereinander oder nebeneinander, besteht auch die Möglichkeit, nur die Spalte oder nur die Zeile zu fixieren. Dort steht dann nur ein Dollarzeichen:

```
fx =$B17
```

```
fx =D$12
```


Tipp: Während man die Formel eingibt, kann man mit der Taste F4 die verschiedenen Möglichkeiten durchlaufen: Spalte und Zeile fixieren, nur Zeile fixieren, nur Spalte fixieren, gar nichts fixieren.

Funktionen und Parameter


In den vorherigen Kapiteln haben wir Excel-Funktionen kennengelernt. Oder besser gesagt: einen Bruchteil der Excel-Funktionen.

Doch woher weiß man, was Excel zu bieten hat, welche Funktion man benutzen muss? Klar, eine Internetsuche wird einem womöglich weiterhelfen. Doch auch Excel selbst bietet eine gute Übersicht über die Funktionen.


Man kann entweder auf den fx-Schriftzug vor der Funktionseingabe klicken. Oder man wählt im Menüband **FORMELN** den Punkt **FUNKTION EINFÜGEN** aus.



Es öffnet sich ein Fenster, in dem man zum einen über Stichworte die Funktion suchen kann. Zum anderen kann man sich durch die Kategorien und die darin enthaltenen Funktionen arbeiten. Darunter gibt Excel den Aufbau der gewählten Funktion und eine kurze Beschreibung aus.



Hat man sich für eine Funktion entschieden, kommt man mit "OK" auf die nachfolgende Seite. Man kann nun hier einen Wert eingeben oder aber in dem Excel-Blatt den Wert oder einen Zellbereich auswählen (das Fenster bleibt dabei geöffnet). Und wie man sieht, zeigt Excel sogar gleich das Ergebnis an.



Mehr als ein Bereich

Nun steht anfangs geschrieben, dass Excel bei Funktion wie MITTELWERT einen Zellbereich als Parameter annimmt, also zum Beispiel =MITTELWERT(D7:D13). Das ist aber nur die halbe Wahrheit. Denn wie man erkennen kann, schreibt Excel selber in der Beschreibung "MITTELWERT(Zahl1;Zahl2;...)". Und auch in dem zweiten Fenster oben gibt es mehrere Eingabefelder. Man hat zwar schon mehrere Werte in einem Bereich, aber man kann auch gleich mehrere Bereiche auf einmal berücksichtigen. Alle werden durch ein Semikolon getrennt.

Hier wird der Mittelwert von 14 Zellen – sofern in allen ein Wert steht – ermittelt:

```
fx =MITTELWERT(A1:A5;A10;B7:B10;C7:C10)
```

Gleiches gilt dann auch für die anderen Funktionen wie SUMME, MAX, ODER ...

Insbesondere bei der Summe spart man sich die Wiederholungen der Summen-Funktion. Die beiden Eingaben sind identisch:

```
fx =SUMME(B3:B8)+SUMME(C3:C8)+SUMME(F3:F8)
```



```
fx =SUMME(B3:C8;F3:F8)
```

Es muss übrigens nicht immer eine Zelle oder ein Zellbereich sein, auch Einzelwerte sind erlaubt:

```
fx =MITTELWERT(C4:C7;10)
```

Erste Hilfe

Ist man sich nicht mehr sicher, wie die Funktion heißt oder welche Parameter sie erwartet, fängt man einfach an zu tippen und Excel zeigt die möglichen Funktionen oder die Parameter an; letzteres, sobald die erste Klammer geöffnet wurde.



Die ausführliche Hilfe


Wenn es etwas ausführlicher sein soll, dann kann man sich die Hilfeseite von Excel einblenden bzw. auf die Homepage von Excel führen lassen.

Eine Möglichkeit ist, nach der Eingabe der Funktion, z. B. =WENN, die F1-Taste zu drücken. Es spielt keine Rolle, ob schon weitere Parameter eingegeben wurden.

Die zweite Möglichkeit ist, eine Funktion auf die Weise einzufügen, wie sie am Anfang des Kapitels beschrieben wurde. Dann sieht man bei der Eingabe der Werte links unten einen Link zur Online-Hilfe.

Errechneten Wert fest einfügen

Copy and paste, kopieren und einfügen kann jeder. Aber einfügen ist nicht gleich einfügen. Und Excel bietet eine ganze Reihe von Möglichkeiten, um einzufügen. Dazu kopiert man eine Zelle oder einen Bereich, wie sonst auch, und klappt dann im Menüband *START* den Punkt *EINFÜGEN* auf (kleiner Pfeil nach unten).



Die meisten Optionen sind dafür da, die ursprüngliche Formatierung beizubehalten, oder eben auch nicht. Auf eine Einfüge-Möglichkeit möchte ich näher eingehen, auf das Einfügen von Werten:



Dazu gleich ein Beispiel:

	A	B	C
1	Wert 1	Wert 2	
2	15	30	
3			
4	Summe:	45	(Formel)
5	fest:		(Wert)

Nun wird die Zelle B4 kopiert, zum Beispiel B5 wird ausgewählt und über diese beschriebene Funktion wird nur der Wert eingefügt. Schaut man sich nun einmal B5 an, wird man feststellen, dass dort gar keine Formel mehr hinterlegt ist, sondern lediglich der Wert, der in B4 errechnet wurde. Selbst wenn man nun die Zahlen oben ändert, bleibt B5 immer gleich.

Und mehr noch, man kann den Wert an gleicher Stelle einfügen. Das heißt, man kann B4 kopieren, ausgewählt lassen und dort wird der Wert eingefügt. Die Formel ist auf ewig verschwunden.

Und wofür genau soll das gut sein? Manchmal spielt es gar keine Rolle mehr, wie genau ein Wert zustande gekommen ist. Was zählt, ist lediglich das Resultat. Wenn man zum Beispiel zu einem Produkt aus einer anderen Tabelle ergänzende Daten per *SVERWEIS* geholt hat, wie auf Seite 19 besprochen, dann kann man diese auch ebenso gut fest übernehmen.

Es könnte auch sein, dass sich die Berechnung auf (Hilfs-)spalten bezieht, die später gar nicht mehr dort sind, weil sie nicht mehr gebraucht werden. Es würde zu einem Fehler kommen. Ansonsten könnte man diese Spalten nicht mehr löschen, ohne dass die Berechnung ungültig wird.

Und es hat Performance-Gründe: Insbesondere bei komplexen Listen muss Excel nach jeder Änderung alle Formeln neu berechnen. Diese Zeit kann man sich sparen, wenn nur noch ein Wert hinterlegt ist. Auch nimmt ein fester Wert weniger Speicherplatz weg als eine Formel.

Automatische Berechnung

Vielleicht ist dem einen oder anderen schon einmal aufgefallen, dass Excel gar nicht neu berechnet. Man zieht zum Beispiel eine Zelle mit einer Formel nach unten, aber es erscheint überall der gleiche Wert.


	A	B	C
1	Wert 1	Wer 2	Summe
2	4	5	9
3	2	3	9
4	7	1	9

Und der aufmerksame Excel-Nutzer wird noch das (anklickbare) Wort "Berechnen" unterhalb der Register sehen.



Schuld daran ist, dass die automatische Arbeitsmappenberechnung wohl deaktiviert wurde. Excel bietet diese Möglichkeit, um bei wirklich umfangreichen Tabellen Rechnerleistung und Zeit zu sparen. Ansonsten berechnet Excel bei jeder Änderung die komplette Arbeitsmappe neu. Meiner Ansicht nach spielt diese Deaktivierung aber für die Dateien, mit denen wir üblicherweise arbeiten, keine Rolle.

Und die Abhilfe: Entweder die Taste F9 drücken, um eine einmalige Berechnung anzustoßen. Oder die Arbeitsmappenberechnung wieder auf "Automatisch" schalten. Das passiert in den Excel-Optionen auf der Seite *FORMELN*.



Einheiten in einer Zelle anzeigen


Viele wissen, dass man mit einem Klick statt einer Zahl einen Betrag mit Wahrung anzeigen kann. Excel bietet dazu den folgenden Button:




Und schon wird 6,23 als 6,23 € in der gleichen Zelle angezeigt. ber das Fenster *ZELLEN FORMATIEREN* kann man nahezu jedes andere Wahrungssymbol aussuchen.

Doch es geht auch mit jedem anderen Text und das bedeutet, mit jeder beliebigen Einheit. Dazu bedient man sich der benutzerdefinierten Formatierung.


In Schritt 1 Nachkommastellen und Tausendertrennzeichen festlegen (dient als Vorbereitung, damit es gleich vorgeblendet wird):



In Schritt 2 wechselt man zu "Benutzerdefiniert":



Und in Schritt 3 erganzt man die Einheit (bitte in Anfuhrungszeichen):



Damit wird in einer einzigen Zelle 1,23 kg angezeigt. Das Schone daran ist, dass es sich wirklich nur um eine Anzeige handelt. Man kann problemlos mit dieser Zelle weitere Berechnungen anstellen.

Bei der Ausgabe hat es den Vorteil, dass man nur eine einzelne Zelle benotigt. Der Nachteil ist jedoch, dass in bestimmten Fallen die Optik etwas auf der Strecke bleibt, als wenn man die Einheit separat schreibt, wie hier zu sehen ist:

	A	B	C	D	E	F
1	Gewicht:	3,25 kg		Gewicht:	3,25	kg
2	Kosten:	2,20 EUR/kg		Kosten:	2,20	EUR/kg
3	ergibt:	5,55 EUR		ergibt:	5,55	EUR

Zirkelbezüge

Bei etwas komplexeren Tabellen oder Berechnungen kann es schon einmal sein, dass Excel auf einen Zirkelbezug aufmerksam macht. Nun, das ist nichts anderes, als dass sich zwei Zellen auf sich selbst beziehen. Zum Beispiel benötigt die Berechnung in Zelle B7 den Wert aus Zelle C9; C9 wiederum möchte den Wert aus B7 haben. Um nicht bei der Berechnung in einer Endlosschleife gefangen zu bleiben, prüft Excel die Abhängigkeiten vorher und gibt eine entsprechende Meldung aus.

Zur Lösung des Problems muss man prüfen, welche Zellen in Abhängigkeit stehen. Es könnte auch sein, dass mehrere Berechnungen nacheinander ausgeführt werden und erst die vierte Zelle einen Zirkelbezug hervorruft.

Glücklicherweise ist es jedoch wahrscheinlicher, dass man lediglich die falsche Zelle ausgewählt hat, anstatt dass es sich um ein logisches Problem in der Aufgabenstellung selbst handelt.


Pivot-Tabellen

Pivot-Tabellen dienen uns dazu, auf schnelle Art und Weise Daten aus einer Tabelle umzustellen und nach anderen Kriterien aufzubauen. Erst einmal gibt es eine Beispieltabelle. Sie ist hier deutlich gekürzt dargestellt, also bitte nicht wundern, wenn später andere Werte auftauchen. Hier zählt eher die Struktur, nicht der Inhalt.


	A	B	C	D	E	F	G
1	Kunde	Land	Vertreter	Umsatz Q1	Umsatz Q2	Umsatz Q3	Umsatz Q4
2	Kunde001	Deutschland	Vertreter E.	44.800	40.700	76.200	48.700
3	Kunde002	Deutschland	Vertreter F.	725.800	874.700	970.700	865.200
4	Kunde003	Deutschland	Vertreter D.	8.500	0	7.300	0
5	Kunde004	Italien	Vertreter C.	0	35.700	21.900	15.300
6	Kunde005	Deutschland	Vertreter A.	1.900	8.500	600	7.300
7	Kunde006	Dänemark	Vertreter F.	41.000	56.700	15.100	55.900
8	...						

Wichtig für die Pivot-Tabelle ist, dass die Daten in einer solchen Form alle gleichmäßig aufgebaut sind. Auch wenn es manchmal gerne gewünscht ist: Es ist nicht möglich, ein Wildwuchs von Daten hiermit zu ordnen. Eine weitere Voraussetzung ist, dass alle Spalten mit einer Überschrift versehen sind.

Nun werden die Spalten A bis G markiert und über den nachfolgenden Button eine Pivot-Tabelle eingefügt. In dem nächsten Fenster einfach "OK" drücken, damit die Pivot-Tabelle in einem neuen Blatt eingefügt wird.




In dem leeren Tabellenblatt wird nun der Platzhalter für die Pivot-Tabelle eingefügt:



Wie dort steht, soll dieser Bereich angeklickt werden. Es öffnet sich ein Bereich auf der rechten Seite, in dem alle Pivot-Felder, nämlich die Spalten aus unserer ursprünglichen Tabelle, so wie Felder für Filter, Spalten, Zeilen und Werte aufgelistet sind.


Nun bitte von den Pivot-Feldern das Land in den Bereich Zeilen ziehen und Kunde in Werte:



Excel erzeugt diese Pivot-Tabelle:

3	Zeilenbeschriftungen	Anzahl von Kunde
4	Belgien	1
5	Dänemark	4
6	Deutschland	61
7	Frankreich	8
8	Italien	8
9	Niederlande	8
10	(Leer)	
11	Gesamtergebnis	90

Als nächstes wird der Vertreter in den Bereich Filter gezogen:




Excel bietet jetzt oberhalb der bisherigen Pivot-Tabelle einen Filter an. In diesem Fall kann man sich zu einem Vertreter die Länder und die Anzahl der Kunden pro Land anzeigen lassen:

	A	B
1	Vertreter	(Alle)
2		Suchen
3	Zeiler	
4	Belgie	<input checked="" type="checkbox"/> (Alle)
5	Dänel	<input checked="" type="checkbox"/> Vertreter A.
6	Deute	<input checked="" type="checkbox"/> Vertreter B.


Auf diese Weise kann man alle Daten aus der ursprünglichen Liste in anderer Form schnell neu aufbauen.

Die einzelnen Felder auf der rechten Seite haben genügend Platz und können auch mehrere Werte aufnehmen, die anschließend ordentlich gruppiert angezeigt werden:




3	Zeilenbeschriftungen ▾
4	▫ Vertreter A.
5	Dänemark
6	Deutschland
7	Italien
8	▫ Vertreter B.
9	Belgien

Jetzt soll es aber nicht nur bei der Anzahl der Kunden bleiben. Wir haben in der ursprünglichen Liste die Umsätze stehen. Bitte die Pivot-Auswahl folgendermaßen aufbauen: den Vertreter in den Filter, erst das Land und dann den Kunden in die Zeilen und alle Umsätze in die Werte. Excel fängt nun auch erst einmal mit der Anzahl an. Bei einem Klick auf den Umsatz erscheint ein Menü. Uns interessieren hier die *WERTFELDEINSTELLUNGEN*:



In dem nachfolgenden Fenster lässt sich die Auswahl von Anzahl auf Summe umstellen (und bei Bedarf ein kürzerer Name vergeben):




Nun – nachdem alle vier Umsätze umgestellt sind – bekommt man eine Liste, mit der man gut arbeiten kann:

2	Vertreter	Vertreter B.	
3			
4	Zeilenbeschriftungen ▾	Summe von Umsatz Q1	Summe von Umsatz Q2
5	▫ Belgien	10900	0
6	Kunde025	10900	0
7	▫ Dänemark	20300	33900
8	Kunde088	20300	33900
9	▫ Frankreich	17500	31000
10	Kunde030	0	17900
11	Kunde031	8500	0
12	Kunde051	0	3200

Das Fenster mit den Wertfeldeinstellungen bietet zudem die Möglichkeit, das Zahlenformat anzupassen. Weitere Gestaltungsoptionen für die gesamte Tabelle gibt es oben in dem Menüband *PIVOTTABLE-TOOLS*.

Aktualisieren

Hat sich einmal an den Basisdaten etwas geändert, wird die Pivot-Tabelle nicht automatisch aktualisiert. In diesem Fall muss einmal der folgende Button gedrückt werden:



AUS DER PRAXIS

An dieser Stelle möchte ich einige Aufgabenstellungen, die sich aus der täglichen Arbeit ergeben, sowie meine Lösung dafür vorstellen. Diesem Kapitel möchte ich zwei Redewendungen voranstellen:

Viele Wege führen nach Rom

Häufig gibt es nicht nur einen Weg, um ein gewünschtes Resultat zu erhalten. Ebenso kann man mit verschiedenen Funktionen zu einer Lösung kommen. Aber manchmal kann man sich auch viele Gedanken machen und eine komplexe Formel aufstellen, obwohl es vielfach gar nicht so kompliziert sein muss. Es reicht vielleicht, eine Hilfsspalte einzufügen, die Tabelle zu filtern oder zu sortieren und die Hilfsspalte anschließend wieder zu löschen. Dabei stellt sich die Frage: Muss die Tabelle nur einmalig bearbeitet werden oder wird ein bestimmtes Kriterium für spätere Filterungen gebraucht?

Übung macht den Meister

Wie kommt man nur auf so etwas? Diese Frage wird häufig gestellt. Formeln lernen ist die eine Sache; sie anzuwenden, eine andere. Man muss so in etwa wissen, was Excel kann und wo die Grenzen sind. Bei der täglichen Arbeit kommt hinzu, dass man weiß, woher man welche Daten (in auswertbarer Form) bekommen kann, um zum Beispiel den Kundenstamm mit den Umsatzdaten zu kombinieren. Häufig ist es leider so, dass man dabei auch mal um die Ecke denken muss. Es ist nicht mit einer simplen Formel getan. Es kommt auf Logik und Kombination und ein bisschen Einfallsreichtum an.

Fremdwährungen umrechnen

xxx

Werte aus einem Text extrahieren

Aus einer anderen Datenbank bekommen wir eine Liste mit Kundennummern und Kundennamen. Allerdings in einer Form, die wir schlecht für weitere Bearbeitungen, zum Beispiel mit dem SVERWEIS, gebrauchen können.

	A	B	C
1	Kunde (komplett)	Kunden-Nr.	Kunden-Name
2	18567 (Meyer)	18567	Meyer
3	01423 (Schmidt)	01423	Schmidt
4	00326 (Ferber)	00326	Ferber

Wir müssten nun die Bestandteile trennen. Wir brauchen hierfür Funktionen der Textbearbeitung, die ab Seite 27 erläutert werden.

Die Kundennummer, die ja immer links steht und immer fünfstellig ist, ist einfach zu ermitteln:

<i>fx</i>	=LINKS(A2;5)
-----------	--------------

Für den Kundennamen benötigen wir die TEIL-Funktion. Sie erwartet drei Parameter: den vollständigen Text, die Position des ersten Zeichens und die Anzahl der Zeichen. Der Kundename allerdings immer unterschiedlich lang. Hier hilft die Funktion LÄNGE. Sie möchte lediglich einen Text als Parameter haben. Die Länge des Kundennamen ist die Länge des gesamten Textes abzüglich 8 (nämlich 5 für die Kundennummer, 2 für die Klammern, 1 für das Leerzeichen). In C2 steht also:

<i>fx</i>	=TEIL(A2;8;LÄNGE(A2)-8)
-----------	-------------------------

Für Zwischendurch noch eine Denksportaufgabe: Der Kundename soll in einer einzigen Zelle extrahiert werden, ohne die Funktion TEIL zu verwenden, nur mit Hilfe der anderen Funktionen von Seite 27.

Variable Länge

Im vorherigen Beispiel war die Länge der Kundennummer immer fix. Was ist aber, wenn die Kundennummer ohne die führenden Nullen kommt, also eine unterschiedliche Anzahl von Zeichen hat?

	A	B	C
1	Kunde (komplett)	Kunden-Nr.	Kunden-Name
2	18567 (Meyer)	18567	Meyer
3	1423 (Schmidt)	1423	Schmidt
4	326 (Ferber)	326	Ferber

In diesem Fall bleibt uns nur, nach einer Gemeinsamkeit zu suchen, eine Stelle, an der alle Kundennummern von Kundennamen getrennt sind. Und das ist das Leerzeichen. Hier hilft uns die Funktion FINDEN. Sie erwartet drei Parameter, wobei der letzte optional ist. Der erste Parameter ist der gesuchte Text und der zweite Parameter der vollständige Text.

Erst einmal nur ein kleiner Test in B2, der uns 6 für das sechste Zeichen in der Zeichenfolge liefert:

<i>fx</i>	=FINDEN(" ";A2)
-----------	-----------------

Mit diesem Wissen können wir die obige Formel korrigieren. In B2:

<i>fx</i>	=LINKS(A2;FINDEN(" ";A2)-1)
-----------	-----------------------------

In C2:

```
fx =TEIL(A2;FINDEN(" ";A2)+2;LÄNGE(A2)-FINDEN(" ";A2)-2)
```

Text in Zahl

Vielleicht ist oben in der ersten Tabelle jemandem aufgefallen, dass Excel die Kundennummern 01423 statt 1423 und 00326 statt 326 liefert. Das kann unter Umständen bei einer nachfolgenden Bearbeitung, zum Beispiel mit dem SVERWEIS, verhängnisvoll sein, denn 01423 ist nun eben nicht 1423.

Der Grund ist, dass Excel nur einen Text aus einem anderen Text zieht. Das Ergebnis ist und bleibt ein Text und keine Zahl. Ebenso in der zweiten Tabelle mit den fehlenden führenden Nullen zieht Excel einen Text heraus. Auch wenn dort die LINKS-Funktion die Nummer 1423 liefert, ist es ein Text. Man kann es auch daran sehen, dass Excel Texte immer linksbündig schreibt und Zahlen immer rechtsbündig – sofern noch keine anderen Formatierungen vorgenommen wurden.

Hier kommt nun die WERT-Funktion zum Einsatz. Somit ergibt sich im ersten Beispiel für die Kundennummer:

```
fx =WERT(LINKS(A2;5))
```

Alles zusammen und gekürzt

Wir nehmen noch einmal als Beispiel die Tabelle mit den Kundennummern variabler Länge. Und in einer weiteren Tabelle der gleichen Datei finden sich Informationen zu den Kunden. Die Kundennummer ist eine Zahl und wir wollen per SVERWEIS den zugehörigen Ort ermitteln. Wie geschrieben, auf die Funktion WERT können wir trotzdem nicht verzichten, da wir keinen Text mit einer Zahl vergleichen können.

	A	B
1	Kunde (komplett)	Ort
2	18567 (Meyer)	Köln
3	1423 (Schmidt)	Berlin
4	326 (Ferber)	München

	A	B
1	146	Frankfurt
2	326	München
3	1423	Berlin
4	4269	Hamburg
5	18567	Köln
6	24612	Stuttgart

Auf eine Hilfsspalte verzichten wir hier. Wir bringen alles in eine Formel in B2 (Tabelle 1):

```
fx =SVERWEIS(WERT(LINKS(A2;FINDEN(" ";A2)-1));Tabelle2!$A:$B;2;FALSCH)
```

Zu der Herleitung dieser Formel gibt es noch ein paar Anmerkungen auf Seite 16.

SVERWEIS mit zwei Indexspalten

Die Funktion SVERWEIS wurde in den Grundlagen auf Seite 19 besprochen und auch im vorherigen Kapitel angewendet. Diese Funktion sucht in einem anderen Bereich nach einem bestimmten Merkmal. Die Betonung müsste hier auf "einem" liegen. Doch wie sieht die Sache aus, wenn man nun nach mehreren Kriterien suchen muss, in den nachfolgenden Tabellen beispielsweise nach Produkt und Variante? Quasi ein doppelter oder mehrfacher Index?

	A	B	C
1	Produkt	Variante	Menge
2	A	1	500
3	B	1	300
4	B	3	800
5	C	2	300
6	C	4	200
7	D	1	200

	A	B	C	D
1	Produkt	Variante	Größe	Farbe
2	A	1	17	schwarz
3	A	2	19	grau
4	B	1	11	schwarz
5	B	2	12	weiß
6	B	3	15	grau
7	B	4	19	braun
8	C	1	12	weiß
9	C	2	14	schwarz
10	C	3	16	braun
11	C	4	20	grau
12	D	1	11	grau
13	D	2	15	schwarz
14	D	3	16	weiß

SVERWEIS kann auf diese Weise erst einmal nicht helfen. Ich bediene mich hierbei gerne einer Hilfsspalte in beiden Tabellen. Und zwar füge ich irgendwo vor den Daten eine Spalte ein – ich nehme hier einmal vor Spalte C in beiden Tabellen – und verknüpfe die beiden Zellen zu einem neuen Wert.

In diesem Fall fügen wir jeweils in beiden Tabellen in C2 und nachfolgenden Spalten ein:

<i>fx</i>	=A2&"-"&B2
-----------	------------

Die beiden Tabellen sehen nun folgendermaßen aus:

	A	B	C	D
1	Produkt	Variante	Hilfsspalte	Menge
2	A	1	A-1	500
3	B	1	B-1	300
4	...			

	A	B	C	D	E
1	Produkt	Variante	Hilfsspalte	Größe	Farbe
2	A	1	A-1	17	schwarz
3	A	2	A-2	19	grau
4	...				

Und nun haben wir nur noch einen Wert und können den SVERWEIS anwenden. In Tabelle 1:

<i>fx</i>	=SVERWEIS(C2;Tabelle2!\$C:\$E;2;FALSCH)
-----------	-----------------------------------------

In diesem Fall steht zwischen im verketteten Text zwischen dem ersten und zweiten Wert ein Bindestrich (A2&"-"&B2). Excel könnte auch ohne diesen Bindestrich – oder irgendein anderes beliebiges Sonderzeichen – arbeiten und hier dürfte nicht viel passieren. Aber wer weiß das schon? Was ist, wenn wir in einem anderen Fall a) 123 und 4567 und b) 1234 und 567 verknüpfen würden? Richtig, in beiden Fällen würde 1234567 herauskommen und Excel würde in beiden Fällen das erste Vorkommen dieser Folge herausuchen.

Datum aus einem Textformat extrahieren

Aus manch externer Datenquelle bekommt man Datumsangaben in einem Textformat geliefert. Aber mit dem Wissen aus den Übungen von Seite 51 können wir die einzelnen Bestandteile herausnehmen und in eine Datumsformatierung übernehmen.

	A	B
1	20181022	(JJJMMTT)
2	05102018	(TTMMJJJJ)

DATUM heißt die Funktion zum Umwandeln und erwartet in dieser Reihenfolge Jahr, Monat und Tag. Wir verwenden die bekannten Funktionen LINKS, RECHTS und TEIL. Man muss natürlich wissen, wie die ursprüngliche Datumsangabe formatiert ist.

In A1 ist es Jahr, Monat und Tag. Somit kommt in C1:

```
fx =DATUM(LINKS(A1;4);TEIL(A1;5;2);RECHTS(A1;2))
```

Und in A2 steht Tag, Monat und Jahr. Somit kommt in C2:

```
fx =DATUM(RECHTS(A2;4);TEIL(A2;3;2);LINKS(A2;2))
```

Übrigens funktioniert die DATUM-Funktion auch, wenn das ursprüngliche Datum eine Zahl ist, kein Text. In C4 kommt die gleiche Formel wie in C1. Excel wandelt intern die Zahl in einen Text um.

4	20181022	(JJJMMTT als Zahl)
---	----------	--------------------

```
fx =DATUM(LINKS(A4;4);TEIL(A4;5;2);RECHTS(A4;2))
```

Im letzten Fall muss das Datum mit dem Jahr beginnen. Würde es mit dem Tag beginnen, könnte es sein, dass das Datum eine Stelle weniger hat (wegen der weggelassenen Null am Anfang). Hier müsste man mit etwas mehr Aufwand die Position bzw. Länge anhand der gesamten Länge der Zahl ermitteln. Wer möchte, darf gerne einmal die Formel nachvollziehen:

6	5102018	(TTMMJJJJ als Zahl; 5.10.2018)
---	---------	--------------------------------

```
fx =DATUM(RECHTS(A6;4);TEIL(A6;2+(A6>10000000);2);LINKS(A6;1+(A6>=10000000)))
```

Hat man die Datumsangabe erst einmal in einer Form als Zahl vorliegen, mit der Excel auch arbeiten kann, lässt sie sich zum Beispiel auch mit einer neuen Formatierung wie JJJJ-MM-TT darstellen.

Duplikate und fehlende Einträge

Möchte man eine große Datenmenge in einer oder auch in mehreren Tabellen abgleichen, hilft es häufig schon, die Anzahl der Einträge zu zählen.

Duplikate

	A	B	C
1	Artikel	Anzahl	mehrfach
2	B	2	ja
3	C	3	ja
4	A	1	nein
5	D	1	nein
6	B	2	ja
7	C	3	ja
8	C	3	ja

In diesem Fall zählen wir einfach durch, wie oft in der gesamten Spalte der Wert der jeweiligen Zeile vorkommt:

```
fx =ZÄHLENWENN(A:A;A2)
```

Die Verwendung der WENN-Funktion sollte mittlerweile kein Problem mehr darstellen. Wenn die genaue Anzahl des Suchergebnisses nicht unbedingt benötigt wird, dann kann man auch schreiben:

```
fx =WENN(ZÄHLENWENN(A:A;A2)>1;"ja";"nein")
```

Die Duplikate sind nun erst einmal auffindig gemacht und können über einen Filter ausgewählt werden.

Fehlende Einträge

Ebenso verhält es sich, wenn man wissen möchte, ob ein Eintrag aus der zweiten Tabelle in der ersten Tabelle vorhanden ist, oder auch nicht.

	A	B	C
1	Artikel	Anzahl	fehlt
2		in Tabelle 1	in Tabelle 1
3	A	1	nein
4	B	2	nein
5	C	3	nein
6	D	1	nein
7	E	0	ja

Die Abfragen funktionieren nun ebenso, wir beziehen uns allerdings auf die erste Tabelle:

```
fx =ZÄHLENWENN(Tabelle1!A:A;A3)
```

```
fx =WENN(ZÄHLENWENN(Tabelle1!A:A;A3)=0;"ja";"nein")
```

Excel ist übrigens in der Lage, doppelte oder eindeutige Einträge in einer Liste optisch darzustellen. Das Stichwort heißt *BEDINGTE FORMATIERUNG* und befindet sich im *START*-Menüband. Aber das ist ein anderes Thema.

Fehler abfangen

Es ist nicht so wirklich schön, wenn Excel in einer Zelle Fehler wie #NV ausgibt. Benutzerfreundlicher wäre es, wenn dort eine aussagekräftige Fehlermeldung steht. Oder je nach Einsatzfall ein Standardwert.

Wie wir gleich sehen werden, gibt es auch hier nicht nur einen Weg, um Fehler abzufangen. Natürlich kann man viel Aufwand betreiben, um eine möglichst fehlerfreie Tabelle zu erzeugen. Grundsätzlich sollte man sich aber vorab kurz darüber Gedanken machen, inwiefern die Fehlerabfrage notwendig ist. Kommen die Daten aus einer zuverlässigen externen Quelle? Wie wahrscheinlich ist es, dass dort zum Beispiel ein Text steht, wo eine Zahl erwartet wird? Und wer ist die Zielgruppe meiner Excel-Tabelle? Muss ich davon ausgehen, dass auch (unwissentlich) Falscheingaben getätigt werden?

Zunächst einmal eine Übersicht, welche Fehler Excel meldet:

Nr.	Fehler	Beschreibung
1	#NULL!	Es gibt keine gemeinsame Schnittmenge von zwei Zellbereichen; wahrscheinlicher ist es, dass der Zellbereich nicht richtig eingegeben wurde. Beispiel: =SUMME(A1:A3 B1:B3)
2	#DIV/0	Eine Division durch null ist selbstverständlich nicht erlaubt. Beispiel: =14/0, oder auch die MITTELWERT-Funktion mit einem leeren Bereich, denn auch dort wird durch die Anzahl 0 geteilt.
3	#WERT!	Eine Zahl wurde erwartet, aber ein Text übergeben. Beispiel: =SUMME(3;5;"sieben";9)
4	#BEZUG!	Der Bezug auf eine(n) Zelle/Zellbereich ist ungültig; wahrscheinlich wurden Zeilen oder Spalten gelöscht, auf die verwiesen wurde.
5	#NAME?	Wahrscheinlich wurde eine Funktion nicht richtig geschrieben. Beispiel: =TEILERGEBNIS(9;A2:A100)
6	#ZAHL!	Ein Wert ist keine gültige Zahl, z. B. zu groß oder falsch formatiert oder nicht erwartet. Beispiel: =WURZEL(-4)
7	#NV	Nicht vorhanden; kommt üblicherweise vor, wenn ein SVERWEIS keinen Eintrag findet.

Fehler vermeiden

Ohne spezielle Formeln können Fehler möglicherweise vorher schon vermieden werden, indem man sich überlegt, welche "Falscheingaben" in einer Excel-Tabelle vorkommen könnten.

Könnte eine Zahl in einer Liste null sein oder könnte der Benutzer null eingeben?

```
fx =WENN(A2=0;"---";A1/A2)
```

Oder existiert der gesuchte Eintrag in der Liste überhaupt?

```
fx =WENN(ZÄHLENWENN(Tabelle2!$A:$A;A2)>0;SVRWEIS(A2;Tabelle2!$A:$C;3;FALSCH);  
A2&" existiert nicht")
```

Fehler abfragen

Um Fehler jeglicher Art abzufragen, bietet Excel die Funktion ISTFEHLER. Damit könnte man auf die Schnelle in der Tabelle sichtbar machen, dass etwas nicht stimmt. Die Funktion liefert WAHR oder FALSCH, was wiederum abgefragt werden kann:

```
fx =WENN(ISTFEHLER(A1/A2);"Eingabefehler!";A1/A2)
```

Ferner bietet Excel die Funktion FEHLER.TYP, die genauer erläutert, welcher Fehler aufgetreten ist. Schauen wir noch einmal in die Tabelle oben; dort steht in der ersten Spalte eine Fehlernummer. Genau diese liefert die Funktion FEHLER.TYP zurück.

Zugegeben, das klingt schon ein wenig nach Programmierung. Es ergibt sicherlich Sinn bei Tabellen mit vielen Benutzereingaben, wenn man dem weniger erfahrenen Benutzer verständliche Fehlermeldungen zurückgeben möchte. Man könnte die Meldungen in eine zweite Tabelle auslagern und von dort abrufen:

	A	B
1	Umsatz:	125478
2	Tage:	21 oder 22
3		
4	Meldung:	Bitte eine gültige Zahl eingeben!

	A	B
1	1	Fehler in der Tabelle ... Bitte den Ersteller kontaktieren!
2	2	An dieser Stelle darf keine 0 oder leere Eingabe stehen!
3	3	Bitte eine gültige Zahl eingeben!
4	4	Fehler in der Tabelle ... Bitte den Ersteller kontaktieren!
5	5	Fehler in der Tabelle ... Bitte den Ersteller kontaktieren!
6	6	Fehler in der Tabelle ... Bitte den Ersteller kontaktieren!
7	7	Wert wurde nicht gefunden!

```
fx =WENN(ISTFEHLER(B1/B2);SVERWEIS(FEHLER.TYP(B1/B2);Tabelle2!$A$1:$B$7;2;FALSCH);B1/B2)
```

Kurzform für bestimmte Fehler

Unter Umständen haben wir beim Abfangen von Fehlern die Berechnung mehrmals eingegeben, einmal als Bedingungssteil einer WENN-Funktion, und im Erfolgsfall noch einmal im Dann-Teil der WENN-Funktion:

```
fx =WENN(ISTFEHLER(A1/A2);"Eingabefehler!";A1/A2)
```

Es geht jedoch auch kürzer, nämlich mit der Funktion WENNFehler. Der zweite Parameter wird nur ausgegeben, wenn die Bedingung einen Fehler liefert, ansonsten das Ergebnis der Bedingung:

```
fx =WENNFehler(A1/A2;"Eingabefehler!")
```

Oder auch mit WENNNV, also wenn der SVERWEIS keinen passenden Eintrag finden würde:

```
fx =WENNNV(SVERWEIS(A2;Tabelle2!$A:$C;3;FALSCH);"nicht gefunden!")
```

Inhalte abfragen

In den vorherigen Beispielen haben wir zwei unterschiedliche Ansätze der Fehlerbehandlung durchgeführt: zum einen den Inhalt vorab geprüft, d. h. erst gar nicht erst einen Fehler auftreten lassen, und zum anderen auf die Fehlermeldung von Excel reagiert. Nehmen wir noch einmal das Beispiel eines häufigen Fehlers, der Division durch null.

Hier prüfen wir vorab den Inhalt und führen dann die Berechnung durch. A1/A2 wird nie einen Division-durch-null-Fehler melden, da A2 in der Berechnung nie null ist:

fx =WENN(A2=0;"---";A1/A2)

Und hier lassen wir Excel erst die Berechnung durchführen und schauen, ob ein Fehler auftritt:

fx =WENNFEHLER(A1/A2;"---")

Excel bietet eine ganze Reihe von Funktionen an, die den Inhalt überprüfen und ganz einfach WAHR oder FALSCH zurückgeben. Die Funktionsnamen sind selbsterklärend, oder? Ansonsten gibt es weitere Erklärungen, wie auf Seite 38 beschrieben:

- ISTZAHL
- ISTLEER
- ISTTEXT
- ISTKTEXT
- ISTGERADE
- ISTUNGERADE.

Falscheingaben ignorieren

Manches Mal reicht es einfach nur aus, eine falsche Eingabe durch einen anderen Wert zu ersetzen. Zum Beispiel soll nur ein Wert von 1 bis 10 möglich sein. Bei einem kleineren Wert wird automatisch 1 angenommen, bei einem größeren Wert automatisch 10. Die Funktionen, die wir hier benötigen, sind MIN und MAX. Wir müssen aber nun einmal kurz um die Ecke denken. Denn MIN bedeutet nicht, dass der eingegebene Wert mindestens einem anderen Wert entsprechen muss. MIN bestimmt aus einer Reihe von Werten den kleinsten Wert. Deswegen benötigen wir für die untere Grenze die MAX-Funktion mit den Parametern Eingabewert und 1; es wird die größere der beiden Zahlen genommen. Für die obere Grenze nehmen wir die MIN-Funktion mit den Parametern Eingabewert und 10; es wird die kleinere der beiden Zahlen genommen.

	A	B
1	Eingabe	Korrektur
2	-5	1
3	5	5
4	15	10

Das Ganze lässt sich auch kombinieren:

fx =MAX(1;MIN(A2;10))

Absatzplanung

Unsere Aufgabe ist es, eine Absatzplanung zu erstellen. Wir müssen angeben, welche Mengen ein Kunde von uns in welchem Monat beziehen wird. Eine Glaskugel bekommen wir nicht. Aber fangen wir doch einmal mit einem ganz einfachen Hilfsmittel an. In einem ersten Schritt müssen soll die zu liefernde Menge an eine Verpackungseinheit (VPE) angepasst werden.

	A	B	C	D
1		Menge	VPE	Plan-Menge
2	Artikel 1	75	150	150
3	Artikel 2	800	192	960
4	Artikel 3	240	120	240
5	Artikel 4	440	150	450

Jetzt müssen wir feststellen, wie oft eine VPE in die gewünschte Menge passt und planen dann die Menge auf Basis von vollen VPE. Dazu teilen wir die Menge und runden anschließend nach oben, somit bekommen wir die Anzahl der VPE. Die Plan-Menge ist dann die Anzahl der VPE multipliziert mit den VPE. In diesen Beispielen beschäftigen wir uns nicht mit Falscheingaben (negative Zahlen, Division durch null), denn die Formeln werden gleich noch umfangreich genug; das ist in den entsprechenden Kapiteln nachzulesen.

$$fx = \text{AUFRUNDEN}(B2/C2;0)*C2$$

Excel hält für diesen Fall sogar eine eigene Funktion vor, die als Parameter den Wert und die Schrittweite (hier: VPE) bekommen muss:

$$fx = \text{OBERGRENZE}(B2;C2)$$

Fast hätten wir es vergessen: Wir haben doch mit den Kunden eine Mindestabnahmemenge vereinbart. Unabhängig davon, ob der Kunde schon eine VPE oder mehr bekommt, soll er mindestens eine bestimmte Menge abnehmen. In einem vorherigen Kapitel wurde schon einmal über MIN und MAX gesprochen; dort ging es zwar um Falscheingaben, aber das Prinzip ist dasselbe (Seite 60). In der letzten Zeile des nachfolgenden Beispiels ist die Mindestmenge kein Vielfaches der VPE, trotzdem soll immer ein Vielfaches geliefert werden.

	A	B	C	D	E
1		Menge	VPE	Min-Menge	Plan-Menge
2	Artikel 1	75	150	300	300
3	Artikel 2	800	192	576	960
4	Artikel 3	240	120	120	240
5	Artikel 4	440	150	500	600

Wir schreiben als neue Planmenge:

$$fx = \text{OBERGRENZE}(\text{MAX}(B2;D2);C2)$$

Das war bis hier doch noch recht einfach, oder? Nun möchten wir aber das ganze Jahr auf dieser Basis durchplanen, d. h. wir verteilen eine Jahresmenge im ersten Schritt erst einmal gleichmäßig auf die Monate. Wir geben nur die drei Werte in der ersten Zeile vor, den Rest soll Excel für uns übernehmen:

	A	B	C	D	E	F	G	H	I	J	K	L
1	Jahresbedarf:	2000	VPE:	108	Min.:	216						
2	Jan.	Feb.	März	April	Mai	Juni	Juli	Aug.	Sept.	Okt.	Nov.	Dez.
3	216	216	216	216	0	216	216	216	0	216	216	216
4	216	432	648	864	864	1080	1296	1512	1512	1728	1944	2160

In vielen Fällen wird – bedingt durch die Vorgaben – die tatsächliche Jahresmenge größer sein. Ansonsten müssten wir im Dezember nur einen Bruchteil liefern. In der untersten Zeile wird lediglich die Plan-Menge zur Information kumuliert; diese Zeile fließt in keine Berechnung ein.

Einen Sonderfall nimmt der Januar ein, weil wir nicht auf die vorherigen Monate Rücksicht nehmen müssen. Wir rechnen wie im Beispiel oben, allerdings nur mit 1/12 der Jahresmenge:

fx = OBERGRENZE(MAX(\$C\$1/12;\$G\$1);\$E\$1)

Im Februar schreiben wir – und bitte nach rechts ausfüllen:

fx =--(AUFRUNDEN((SUMME(\$A\$3:A3)-(SPALTE()-1)*\$C\$1/12);0)<=(\$C\$1/12))*
OBERGRENZE(MAX(((C\$1/12*SPALTE())-SUMME(\$A\$3:A3));\$G\$1);\$E\$1)

Okay, das bedarf wohl jetzt einer Erklärung. Im Januar konnten wir noch ein Jahresbedarf nehmen, durch zwölf teilen und auf die nächste VPE oder Mindestmenge aufrunden. Nun müssen wir in den nachfolgenden Monaten aber aufpassen. Der Monatsdurchschnitt beträgt in diesem Fall 167 Stück (2000 Stück/12). Nach diesem Schema würden wir jedes Mal 216 Stück einplanen. Bereits nach vier Monaten hätten wir den Bedarf von fünf Monaten geplant. Deswegen müssen wir einen Monat aussetzen. Es darf nicht pauschal die Mindestmenge angenommen werden. Aus dem Grund muss verglichen werden, welche Menge bereits verplant ist und welche Menge rein theoretisch auf Basis der Summe der Durchschnittswerte noch benötigt wird. Insbesondere bei hohen Liefermengen und kleinen Bedarfen ist möglicherweise der Bedarf für mehrere Monate gedeckt.

Aber vorab ein paar Anmerkungen zu der Formel. Warum diese merkwürdige Schreibweise SUMME(\$A\$3:A3)? Nur wegen des Komforts. Damit lässt sich die Formel einfach nach rechts ziehen. Wir erhalten immer die Summe von der fixierten Zelle A3 (Januar) bis zum Vormonat. Wofür wird =SPALTE() verwendet? Es wird die Nummer der Spalte zurückgegeben, in der diese Formel steht. Steht die Formel beispielsweise in D4, dann ist =SPALTE() dort 4, und somit der April. Bleiben noch die beiden Minuszeichen -- am Anfang. Wie man sieht, steht in dem Teil dahinter (ab der ersten geöffneten Klammer) ein Vergleich. Ein Vergleich liefert einen Wahrheitswert (WAHR oder FALSCH) zurück. Excel bietet mit -- vorweg die Möglichkeit, aus einem Wahrheitswert eine Zahl zu generieren. --WAHR ergibt 1 und --FALSCH ergibt 0. Eine gleichwertige Alternative ist die Funktion =N(). Die Rundung am Anfang musste ich einbringen, da es ansonsten in bestimmten Fällen durch die interne Berechnung mit Nachkommastellen zu einer zusätzlichen Lieferung im Dezember käme.

Damit wäre die Formel technisch erklärt. Aber was passiert nun genau? In dem ersten Teil bis vor OBERGRENZE prüfe ich ganz einfach, ob wir schon mehr geplant haben als die summierten Durchschnittswerte in dem Zeitraum bis zur aktuellen Formel. Das bedeutet nämlich, wie weiter oben erklärt, dass wir einen Monat aussetzen müssen. In dem Fall ist das Ergebnis aus der ersten Hälfte der Formel 0. Und 0 multipliziert mit dem ganzen Rest bleibt bei 0. Im anderen Fall errechnen wir die Differenz zwischen der Menge, die tatsächlich geplant ist, und dem Durchschnitt in diesem Zeitraum, angepasst an VPE und Mindestmenge.

Wir haben bei einem Produkt ein Saisongeschäft oder möchten unsere Urlaubszeiten oder die des Kunden berücksichtigen? Dann arbeiten wir einfach mit einem Verteilschlüssel. Es spielt keine Rolle, ob wir nun Prozentzahlen, Arbeitstage oder einen beliebigen Verteilerschlüssel verwenden. Ich entscheide mich dafür, einem "normalen" Monat den Wert 10 zu geben und alle anderen Monate ins Verhältnis zu setzen.

	A	B	C	D	E	F	G	H	I	J	K	L
1	Jahresbedarf:	2000	VPE:	108	Min.:	216						
2	Jan.	Feb.	März	April	Mai	Juni	Juli	Aug.	Sept.	Okt.	Nov.	Dez.
3	216	216	216	216	216	0	216	216	216	0	216	216
4	216	432	648	864	1080	1080	1296	1512	1728	1728	1944	2160
5	Verteilung:											
6	10	9	10	7	10	10	5	8	10	10	10	5

Im vorherigen Beispiel wurde in jedem Monat der Bedarf mit "/12" gleichmäßig aufgeteilt. Wenn wir mit Verteilerschlüssel arbeiten, multiplizieren wir den Jahresbedarf mit dem Verteilerwert in dem Monat bzw. der Summe der Monate und teilen durch die Summe aller Monate. Das sieht dann im Januar so aus:

fx =OBERGRENZE(MAX(\$C\$1*A6/SUMME(\$A\$6:\$L\$6);\$G\$1);\$E\$1)

Und im Februar bzw. fortgeführt in allen weiteren Monaten:

fx =--(SUMME(\$A\$3:A3)<AUFRUNDEN(\$C\$1/SUMME(\$A\$6:\$L\$6)*SUMME(\$A\$6:B6);0))
*OBERGRENZE(MAX((((\$C\$1/SUMME(\$A\$6:\$L\$6)*SUMME(\$A\$6:B6))-SUMME(\$A\$3:A3))
;\$G\$1);\$E\$1)

Zugegeben, eine solche Berechnung muss man sich mit etwas Mühe erarbeiten. Aber ist es nicht schön, alles in einer Formel unterzubringen, ohne Hilfsspalte, ohne Programmierung?

Nun sind wir in der Lage, eine Vorschau für alle unsere Kunden und Artikel zu erstellen (beim Erarbeiten genau auf die Fixierung mit \$ achten, damit das Kopieren der Formeln möglichst einfach wird):

	A	B	C	D	E	F	G	H	I
1			Arbeitstage:			22	20	23	18
2	Kunde	Artikel	VPE	Min.-M.	J-Menge	Jan.	Feb.	März	April
3	Berger	KM256	120	360	12000	1200	960	1200	840
4	Berger	QD573	150	150	4500	450	450	450	300
5	Berger	BJ696	256	512	7300	768	768	512	512
6	Konradi	AL412	108	324	3200	324	324	324	324
7	Konradi	KM256	120	240	6000	600	480	600	480
8	Schultz	QD573	150	150	4000	450	300	450	300
9	Schultz	MM489	72	144	1600	216	144	144	144

	J	K	L	M	N	O	P	Q	R
1	21	20	14	23	21	20	21	15	
2	Mai	Juni	Juli	Aug.	Sept.	Okt.	Nov.	Dez.	Summe
3	1080	1080	600	1200	1080	960	1080	720	12000
4	450	300	300	450	300	450	450	150	4500
5	768	512	512	768	512	768	512	512	7424
6	324	324	0	324	324	324	324	0	3240
7	480	600	240	600	600	480	480	360	6000
8	300	300	300	450	300	300	300	300	4050
9	144	144	0	216	144	144	144	144	1728

Gleitender Durchschnitt

Auf Seite 14 hatten wir den Mittelwert kennengelernt. Aus einer Reihe von Werten wird ein Durchschnitt gebildet. Hat man jedoch eine viel größere Menge an Werten, bedient man sich gerne des gleitenden Durchschnitts. Man könnte ihn auch als rollierenden Durchschnitt bezeichnen. Man bekommt dadurch eine Tendenz angezeigt mit – je nach Vorgabe – Vernachlässigung älterer Werte.

Möchte man beispielsweise wissen, wie sich die Gehälter in einer Branche entwickelt haben, könnte man im Jahr 2019 die Mittelwerte von 2014–18 betrachten, im Jahr 2020 die Jahre 2015–19 und im Jahr 2021 die Jahre 2016–20. Manchmal spielen noch ältere Werte für die Durchschnittsberechnung einfach keine Rolle und es gibt Faktoren und Bedingungen, die das Ergebnis verfälschen bzw. bei denen das Ergebnis keine Aussagekraft mehr hat. Ein anderes Beispiel ist die Börse. Hier stehen Datenmengen der letzten Jahre zur Verfügung. Je nach Anlagestrategie interessieren einen die letzten Monate, Wochen oder auch nur wenige Tage.

Excel bietet selbst Tools zu diesen Themen an (Stichwort Add-In Analyse und Trendlinie in Diagrammen), aber im Rahmen dieses Buchs wollen wir uns auf die Formeln konzentrieren.

Ein paar Worte vorweg: Hier stelle ich gleich drei Berechnungsmethoden vor, deren Bezeichnungen ich ebenfalls der Börsen-Charttechnik entnehme. Für die Beispiele und Grafiken nehme ich einhundert Werte, auch wenn die Tabellen nur die letzten Werte wiedergeben. Die Anzeige ist auf eine Nachkommastelle beschränkt.


Einfach gleitender Durchschnitt – simple moving average (SMA)

	A	CM	CN	CO	CP	CQ	CR	CS	CT	CU	CV	CW
1	Periode	P-10	P-9	P-8	P-7	P-6	P-5	P-4	P-3	P-2	P-1	P0
2	Werte	45	48	50	48	51	56	53	55	57	53	54
3	SMA(10)	36,6	38,2	40,5	42,1	43,6	45,7	47,3	48,7	50,4	51,6	52,5

Der einfach gleitende Durchschnitt bedient sich einfach immer der letzten x Werte. In diesem Fall habe ich mich für 10 entschieden. Und die Berechnung ist auch sehr einfach. In der letzten Zelle CW3 steht:

```
fx =MITTELWERT(CN2:CW2)
```

Die Formel wird nach vorne hin ausgefüllt. Die ersten neun Werte bleiben frei, da wir zehn Werte für den einfach gleitenden Durchschnitt benötigen. Das nachfolgende Diagramm veranschaulicht den gleitenden Durchschnitt.



Gewichtet gleitender Durchschnitt – weighted moving average (WMA)

Bei der Betrachtung des vorherigen Diagramms fällt bei dem starken Anstieg am Ende auf, dass der Durchschnitt quasi hinterherhängt. Das liegt daran, dass die letzten zehn Werte ohne Berücksichtigung des Alters voll in die Berechnung einfließen. Stattdessen könnte man nach Belieben diese Werte gewichten. Zum Beispiel soll der neueste Wert (Periode 0) mit 100 % einfließen, der vorherige Wert (Periode -1) nur noch mit 90 % usw.

	A	CM	CN	CO	CP	CQ	CR	CS	CT	CU	CV	CW
1	Periode	P-10	P-9	P-8	P-7	P-6	P-5	P-4	P-3	P-2	P-1	P0
2	Werte	45	48	50	48	51	56	53	55	57	53	54
3	WMA(10)	39,1	41,1	43,3	44,4	46,3	48,5	49,9	51,3	52,8	53,2	53,7

Die Formel in CW3 lautet:

$$fx = (CN2+2*CO2+3*CP2+4*CQ2+5*CR2+6*CS2+7*CT2+8*CU2+9*CV2+10*CW2)/55$$

Eine recht simple Berechnung. Am Ende wird durch 55 geteilt. Das ist die Summe der Gewichtungen.


Man könnte es etwas kürzer schreiben:

$$fx = \text{SUMMENPRODUKT}(CN2: CW2; \{1.2.3.4.5.6.7.8.9.10\})/55$$

Zu dem Summenprodukt gibt es bereits ein eigenes Kapitel auf Seite 25. Hier werden zwei Arrays gleicher Größe übergeben, zum einen der Bereich CN2: CW2 (zehn Werte) und zum anderen ein händisches Array (ebenfalls zehn Werte). Diese werden dann multipliziert und summiert: CN2 mit 1 plus CO2 mit 2 plus CP2 mit 3 usw.

Excel 2021 Mit der neuen Funktion SEQUENZ kann das zweite Array ersetzt werden:

$$fx = \text{SUMMENPRODUKT}(CN2: CW2; \text{SEQUENZ}(1;10))/55$$



Die Kurve ist nun näher an den aktuellen Werten, trotzdem ist sie geglättet und verdeutlicht die Durchschnitte trotz Schwankungen und Ausreißer der Werte.

Exponentiell gewichteter Durchschnitt – exponential moving average (EMA)

Der exponentiell gewichtete Durchschnitt geht einen etwas anderen Weg als die beiden vorherigen. Es wird immer der vorherige Durchschnitt mit einbezogen, der seinerseits den Durchschnitt davor einbezieht usw. Somit entsteht ebenfalls eine stärkere Gewichtung auf die neuesten Werte. Es ist ein beliebter Indikator in der Börsen-Charttechnik. Die nachfolgende Formel findet man im Internet, man muss sie nicht auswendig kennen und sie sieht komplizierter aus, als sie ist:

$$EMA_t = (C_t - EMA_{t-1}) * SF + EMA_{t-1}$$

EMA_t ist der zu errechnende Durchschnitt; EMA_{t-1} der Durchschnitt der vorherigen Periode; C_t der aktuelle Wert und SF ein sogenannter smoothing factor mit der Berechnung $SF = 2 / (\text{Anzahl Perioden} + 1)$.

Wir haben alle Werte vorliegen und brauchen sie nur noch in die Berechnung einfließen lassen. Dazu erst einmal wieder die Tabelle:


	A	CM	CN	CO	CP	CQ	CR	CS	CT	CU	CV	CW
1	Periode	P-10	P-9	P-8	P-7	P-6	P-5	P-4	P-3	P-2	P-1	P0
2	Werte	45	48	50	48	51	56	53	55	57	53	54
3	EMA(10)	37,4	39,4	41,3	42,5	44,1	46,2	47,5	48,8	50,3	50,8	51,4
4	EMA(40)	26,6	27,6	28,7	29,6	30,6	31,9	32,9	34,1	35,1	36,0	36,9

EMA_t ist der zu errechnende Wert in CW3; EMA_{t-1} ist der vorherige Wert aus CV3; C_t ist der aktuelle Wert aus CW2; SF ist das Ergebnis von $2/(10+1)$. Daraus ergibt sich für CW3:

$$fx = (CW2 - CV3) * 2 / (10 + 1) + CV3$$

Zur Veranschaulichung ist in Zeile 4 der exponentiell gewichtete Durchschnitt mit einer Periodenlänge von 40 angegeben. In der Formel für CW4 braucht nur ein Wert geändert werden:

$$fx = (CW2 - CV4) * 2 / (40 + 1) + CV4$$



Bleibt noch die Frage, welche Berechnung man nun nehmen soll. Und mit welchem Zeitraum. Das hängt – wie so häufig – vom Einzelfall ab. Anfangs wurde das Beispiel Börse erwähnt. Einen langen Zeitraum nimmt man, um die langfristige Trendrichtung darzustellen (siehe dazu auch die Grafik der EMA(40): wir liegen aktuell deutlich über dem Durchschnitt der letzten 40 Perioden), einen kurzen Zeitraum mit verstärktem Fokus auf die aktuelleren Kurswerte für den derzeitigen Trend. Der einfach gleitende Durchschnitt ist etwas träger als die anderen beiden Methoden. Das kann allerdings von Vorteil sein, wenn man eine längerfristige Entwicklung darstellen möchte. Es kommt immer darauf an, was man möglichst aussagekräftig damit ausdrücken möchte.

Eine traurige Situation, aber ein passendes Beispiel: Während der Corona-Pandemie wurden die Neuinfektionen teilweise als Sieben-Tage-Mittel dargestellt. Sieben Tage sind eng genug, um die derzeitige Höhe realistisch auszuweisen und alte Werte nicht zu berücksichtigen. Aber sieben Tage sind auch weit genug, um die Schwankungen und Nachmeldungen nach einem Wochenende zu glätten. Anders lag es bei der Sieben-Tage-Inzidenz (Neuinfektionen pro 100.000 Einwohner): Hier wurden die letzten sieben Tage summiert. Das rollierende Prinzip blieb dasselbe.